

Российская академия наук  
Сибирское отделение  
Институт систем информатики  
имени А. П. Ершова

Н.С. Карнаухов, Д.Ю. Першин, В.И. Шелехов

**ЯЗЫК ПРЕДИКАТНОГО ПРОГРАММИРОВАНИЯ Р**

Препринт  
153

Новосибирск 2010

Язык предикатного программирования **P** ориентирован на разработку надежных и эффективных программ в классе задач дискретной и вычислительной математики. По сравнению с начальной версией [1] язык существенно расширен средствами описания процессов, алгебраическими типами, агрегатами, языком представления формул и другими конструкциями. Язык **P** по синтаксису, набору операторов и операций и др. особенностям существенно приближен к стилю языков семейства **C**.

**Siberian Branch of the Russian Academy of Sciences  
A. P. Ershov Institute of Informatics Systems**

**N. Karnaukhov, D. Perchine, V. Shelekhov**

**THE PREDICATE PROGRAMMING LANGUAGE P**

**Preprint  
153**

**Novosibirsk 2010**

The predicate programming language P is oriented on the development of reliable and effective programs for problems in discrete and calculus mathematics. In comparison with the initial version [1] the language is essentially extended by the means for process descriptions, algebraic types, aggregates, formula description language, and another constructs. For the P language, the syntax, operator set, and other features are very similar to those for the C family languages.

## 1. ВВЕДЕНИЕ

Язык предикатного программирования **P** (Predicate programming language) является универсальным и может использоваться для разработки широкого класса программ. По меньшей мере этот класс включает программы (разрабатываемые обычно на языке ФОРТРАН) для задач вычислительной математики. В целях спецификации и реализации программ для реактивных систем язык **P** расширен конструкциями описания процессов, передачи сообщений и порождения процессов, работающих параллельно с процессом-родителем. В меньшей степени язык **P** пригоден для задач системного программирования, где обычно используются объектно-ориентированные средства, которых нет в ядре языка **P**.

По сложившейся классификации язык **P** следует отнести к классу языков функционального программирования. Язык **P** сочетает функциональный и операторный (предикатный) стили записи алгоритмов и обладает существенно большей выразительностью по сравнению с чисто функциональными языками. В этом смысле язык **P** ближе к языкам логического программирования. Однако в отличие от логических языков процесс вычисления программы на языке **P** реализуется не логическим выводом, а явным исполнением, характерным для императивных и функциональных языков.

Язык **P** содержит средства спецификации программы. Программа состоит из набора определений предикатов. Определение предиката имеет вид:

$$A(x: y) \text{ pre } P(x) \{ S \} \text{ post } Q(x, y) ,$$

где  $A$  – имя определяемого предиката,  $x$  – аргументы,  $y$  – результаты,  $S$  – оператор,  $P(x)$  – предусловие и  $Q(x, y)$  – постусловие. Тотальная корректность предиката  $A$  определяется истинностью следующей формулы:

$$P(x) \Rightarrow [LS(S)(x, y) \Rightarrow Q(x, y)] \& \exists y. LS(S)(x, y) \quad (1)$$

где  $LS(S)$  – логическая формула, определяемая для оператора  $S$  в соответствии с логической семантикой  $LS$  [6]. Корректность программы относительно спецификации определяется корректностью каждого определяемого предиката программы. Формулы корректности генерируются по программе автоматически, что позволяет применять для их доказательства системы автоматического доказательства, например такие,

как PVS [7]. Отметим, что для императивных и функциональных языков генерация формул корректности является нетривиальной задачей; кроме того, там намного сложнее исходная спецификация, генерируемые формулы корректности и их последующее доказательство.

Функциональный язык удобнее императивного для разработки алгоритма. Однако получение эффективной программы для функциональных языков проблематично даже с применением изопренной оптимизации в процессе трансляции. Поэтому после отладки программы на функциональном языке для достижения требуемой эффективности ее приходится переписывать на императивный язык. Преимущество технологии предикатного программирования в том, что она является сквозной: к предикатной программе применяется набор трансформаций с получением эффективной программы на императивном расширении языка  $P$  (см. разд. 11), после чего программа может конвертироваться на любой из императивных языков: C, C++, ФОРТРАН и др. Базовыми трансформациями являются:

- склеивание переменных, реализующее замену нескольких переменных одной;
- замена хвостовой рекурсии циклом;
- подстановка определения предиката на место его вызова;
- кодирование структурных объектов низкоуровневыми структурами с использованием массивов и указателей.

C помощью трансформаций можно получить программу предельной эффективности.

Язык предикатного программирования  $P$  впервые представлен в работе [1]. Полное описание языка  $P$  дается в препринте [2]. Последующие модификации и расширения языка инициированы использованием языка для описания различных алгоритмов [3-5]. Введены средства спецификации определений предикатов в виде предусловий и постусловий. Имеется опыт использования языка  $P$  в качестве языка публикации алгоритмов. Язык  $P$  расширен для спецификации и реализации реактивных систем, определяемых в виде совокупности взаимодействующих процессов [5]. Введены средства изображения процессов, операторы приема и отправки сообщений, средства динамического порождения процессов.

Все предыдущие версии языка  $P$  по стилю были ближе к языкам Паскаль и Модуль-2. Последняя версия языка  $P$ , представленная в настоящем описании, по синтаксису, набору операторов и операций и дру-

гим особенностям существенно приближена к стилю языков типа С. Как следствие, язык Р станет более комфортным для программирующих на языках С, С++, Java, С#. Изменения синтаксиса языка Р стали причиной проведения серии серьезных модификаций во всем языке. Определенное влияние на новую версию языка Р оказал язык спецификаций PVS [7]. Введены алгебраические типы. Последовательности заменены списками.

Синтаксис языка Р описывается на расширенном языке Бэкусовских нормальных форм (БНФ) со следующими особенностями:

- терминальные символы выделены жирным шрифтом;
- [ **фрагмент** ] означает возможное отсутствие в синтаксическом правиле заключенного в квадратные скобки **фрагмента**; **фрагмент** определяет последовательность терминальных и нетерминальных символов;
- ( **фрагмент** )\* определяет повторение **фрагмента**; нуль или более раз; круглые скобки могут быть опущены, если **фрагмент** состоит из одного символа
- ( **фрагмент** )+ определяет повторение **фрагмента** один или более раз;
- запись вида [:CLASS:] обозначает символ указанного класса; используются следующие классы символов:
  - alpha – буквенный символ, принадлежащий латинскому или русскому алфавиту; разрешаются заглавные и строчные буквы;
  - digit – цифра (0, 1, 2, 3, 4, 5, 6, 7, 8 или 9);
  - alnum – символ, принадлежащий alpha или digit;
  - blank – пробел или символ табуляции;
  - xdigit – шестнадцатеричная цифра (digit – заглавная или строчная буква от А до F);
  - print – символ, для которого определено начертание (т.е., символ из alnum, blank либо какой-либо другой символ, который можно отобразить).

## 2. ЛЕКЕМЫ

Текст программы представлен в виде последовательности строк символов. Переход на новую строку эквивалентен символу “пробел”. Программа может содержать комментарии, текст которых считается не принадлежащим тексту программы. Комментарий начинается парой символов /\* и завершается парой символов \*/. Второй вид комментариев

начинается с пары символов // и продолжается до конца текущей строки текста. Текст программы составляется из лексем следующего вида:

ЛЕКСЕМА ::= ИДЕНТИФИКАТОР | КЛЮЧЕВОЕ-СЛОВО | КОНСТАНТА |  
ОПЕРАЦИЯ | РАЗДЕЛИТЕЛЬ | МЕТКА

ИДЕНТИФИКАТОР ::= НАЧАЛО-ИДЕНТИФИКАТОРА СИМВОЛ-ИДЕНТИФИКАТОРА\*

НАЧАЛО-ИДЕНТИФИКАТОРА ::= \_ | [:alpha:]

СИМВОЛ-ИДЕНТИФИКАТОРА ::= \_ | [:alnum:]

Идентификатор используется для именованя предикатов, переменных, типов и других объектов. Использование специального имени “\_” в качестве результирующей переменной вызова предиката (см. разд. 3.3) обозначает пустой результат, неиспользуемый в дальнейшем вычислении. Имя “\_” зарезервировано в языке P – его нельзя использовать для именованя объектов программы. Другие идентификаторы, начинающиеся с “\_”, считаются обычными и могут использоваться в описаниях программы.

МЕТКА ::= СИМВОЛ-ИДЕНТИФИКАТОРА\*

КЛЮЧЕВОЕ-СЛОВО ::=

after | array | as | bool | break | case | char | class | else |  
enum | exists | extends | default | false | for | forall |  
formula | if | in | inf | int | import | message | module | nan |  
nat | new | nil | or | post | pre | process | predicate |  
queue | real | receive | set | string | struct | subtype |  
switch | type | true | union | var | while | with | xor

ОПЕРАЦИЯ ::=

+ | - | \* | / | % | ^ | ! | << | >> | ~ | & | | | ? | = | < | >  
| <= | >= | != | => | <=>

РАЗДЕЛИТЕЛЬ ::=

( | ) | [ | ] | { | } | , | ; | : | . | .. | [:blank:]

КОНСТАНТА ::=

ЦЕЛАЯ-КОНСТАНТА  
| ВЕЩЕСТВЕННАЯ-КОНСТАНТА  
| СИМВОЛЬНАЯ-КОНСТАНТА  
| СТРОКОВАЯ-КОНСТАНТА  
| ЛОГИЧЕСКАЯ-КОНСТАНТА  
| КОНСТАНТА-ПУСТО

ЦЕЛАЯ-КОНСТАНТА ::= ЦЕЛОЕ-ЧИСЛО

ЦЕЛОЕ-ЧИСЛО ::= [ЗНАК] ЦИФРЫ | 0x ШЕСТНАДЦАТЕРИЧНЫЕ-ЦИФРЫ

ЗНАК ::= + | -

```

ЦИФРЫ ::= ЦИФРА+
ЦИФРА ::= [:digit:]
ШЕСТНАДЦАТЕРИЧНЫЕ-ЦИФРЫ ::= ШЕСТНАДЦАТЕРИЧНАЯ-ЦИФРА+
ШЕСТНАДЦАТЕРИЧНАЯ-ЦИФРА ::= [:xdigit:]
ВЕЩЕСТВЕННАЯ-КОНСТАНТА ::=
    [ЗНАК] ЦИФРЫ [. ЦИФРЫ] [ПОРЯДОК]
    | [ЗНАК] inf
    | nan
ПОРЯДОК ::= (e | E) [ЗНАК] ЦИФРЫ

```

Для вещественных типов (см. разд. 7) определены специальные значения: **inf** (бесконечность) и **nan** (не число). Значение **nan** возникает в процессе вычисления как результат взятия квадратного корня из отрицательного числа, деления ноля на ноль, умножения ноля на бесконечность и других операций над вещественными числами, когда результат не может быть определён.

```

СИМВОЛЬНАЯ-КОНСТАНТА ::= ' СИМВОЛ '
СИМВОЛ ::= ОБЫЧНЫЙ-СИМВОЛ | СПЕЦИАЛЬНЫЙ-СИМВОЛ |
    ШЕСТНАДЦАТЕРИЧНЫЙ-КОД-СИМВОЛА
ОБЫЧНЫЙ-СИМВОЛ ::= любой символ из [:print:], кроме ' и "
СПЕЦИАЛЬНЫЙ-СИМВОЛ ::= \ " | \ ' | \\ | \0 | \n | \r | \t
ШЕСТНАДЦАТЕРИЧНЫЙ-КОД-СИМВОЛА ::=
    \x ШЕСТНАДЦАТЕРИЧНАЯ-ЦИФРА
    [ШЕСТНАДЦАТЕРИЧНАЯ-ЦИФРА
    [ШЕСТНАДЦАТЕРИЧНАЯ-ЦИФРА [ШЕСТНАДЦАТЕРИЧНАЯ-ЦИФРА]]]

```

Константа типа **char** (см. разд. 7) может быть либо символом класса **print** (см. разд. 1), либо специальной комбинацией символов, заключённой в одинарные кавычки. Специальные комбинации символов могут

Таблица 1

#### Специальные комбинации символов

<code>\''</code>	Двойные кавычки
<code>\'</code>	Одинарные кавычки
<code>\\</code>	Обратная косая черта
<code>\0</code>	Символ с кодом 0
<code>\n</code>	Перевод строки (код 10)
<code>\r</code>	Возврат каретки (код 13)
<code>\t</code>	Символ табуляции
<code>\xNNNN</code>	Символ с шестнадцатеричным кодом NNNN, состоящим не более чем из четырёх цифр.

также встречаться в составе строковых констант.

СТРОКОВАЯ-КОНСТАНТА ::= " СИМВОЛ \* "  
ЛОГИЧЕСКАЯ-КОНСТАНТА ::= **true** | **false**  
КОНСТАНТА-ПУСТО ::= **nil**

### 3. ПРЕДИКАТЫ

Программа состоит из набора определений предикатов.

#### 3.1. Определение предиката

ОПРЕДЕЛЕНИЕ-ПРЕДИКАТА ::= ИМЯ-ПРЕДИКАТА ОПИСАНИЕ-ПРЕДИКАТА  
ИМЯ-ПРЕДИКАТА ::= ИДЕНТИФИКАТОР  
Значением ОПИСАНИЯ-ПРЕДИКАТА является предикат, обозначаемый именем предиката.  
ОПИСАНИЕ-ПРЕДИКАТА ::= ОПИСАНИЕ-ПРЕДИКАТА-ФУНКЦИИ |  
ОПИСАНИЕ-ПРЕДИКАТА-ГИПЕРФУНКЦИИ  
ОПИСАНИЕ-ПРЕДИКАТА-ФУНКЦИИ ::=  
ЗАГОЛОВОК-ПРЕДИКАТА [**pre** ПРЕДУСЛОВИЕ]  
ТЕЛО-ПРЕДИКАТА [**post** ПОСТУСЛОВИЕ]  
ЗАГОЛОВОК-ПРЕДИКАТА ::=  
( [ОПИСАНИЯ-АРГУМЕНТОВ] : ОПИСАНИЯ-РЕЗУЛЬТАТОВ )  
ТЕЛО-ПРЕДИКАТА ::= БЛОК  
ПРЕДУСЛОВИЕ ::= ФОРМУЛА  
ПОСТУСЛОВИЕ ::= ФОРМУЛА

Значения аргументов предиката должны удовлетворять предусловию. По завершении исполнения тела предиката должно быть истинным постусловие, связывающее значения аргументов и результатов.

ОПИСАНИЯ-РЕЗУЛЬТАТОВ ::=  
ИЗОБРАЖЕНИЕ-ТИПА ИМЯ-РЕЗУЛЬТАТА ( , ИМЯ-РЕЗУЛЬТАТА ) \*  
[ , ОПИСАНИЯ-РЕЗУЛЬТАТОВ ]  
ИМЯ-РЕЗУЛЬТАТА ::= ИДЕНТИФИКАТОР | ИДЕНТИФИКАТОР ' ,

Имя вида *имя'* используется для именованного результирующего параметра при условии, что *имя* описано в качестве одного из аргументов с тем же типом. В императивной программе, получаемой трансформацией предикатной программы, результат с именем *имя'* склеивается с соответствующим аргументом *имя*. Если тип аргумента *имя* параметризован (см. разд. 7), то значение параметра для типа результата *имя'* может отличаться.

```

ОПИСАНИЯ-ГРУППЫ-АРГУМЕНТОВ ::=
    ИЗОБРАЖЕНИЕ-ТИПА ИМЯ-АРГУМЕНТА (, ИМЯ-АРГУМЕНТА)* |
    ИЗОБРАЖЕНИЕ-ТИПА-КАК-ПАРАМЕТРА
ОПИСАНИЯ-АРГУМЕНТОВ ::=
    ОПИСАНИЯ-ГРУППЫ-АРГУМЕНТОВ [, ОПИСАНИЯ-АРГУМЕНТОВ]
ИМЯ-АРГУМЕНТА ::= ИДЕНТИФИКАТОР

```

```

1 assign (int from : int to) { to = from }
2 main (list (string) argv : int ret_code) { ret_code = 0 }
3 power (real x, int p : real x') { x' = x^p }

```

Пример 1. Определения предикатов

```

ОПИСАНИЕ-ПРЕДИКАТА-ГИПЕРФУНКЦИИ ::=
    ЗАГОЛОВОК-ПРЕДИКАТА-ГИПЕРФУНКЦИИ [ПРЕДУСЛОВИЯ-ГИПЕРФУНКЦИИ]
    ТЕЛО-ПРЕДИКАТА [ПОСТУСЛОВИЯ-ВЕТВЕЙ]
ЗАГОЛОВОК-ПРЕДИКАТА-ГИПЕРФУНКЦИИ ::=
    ( [ОПИСАНИЯ-АРГУМЕНТОВ] : ОПИСАНИЯ-РЕЗУЛЬТАТОВ-ГИПЕРФУНКЦИИ )
ОПИСАНИЯ-РЕЗУЛЬТАТОВ-ГИПЕРФУНКЦИИ ::=
    [ОПИСАНИЯ-РЕЗУЛЬТАТОВ-ВЕТВИ] [✱ МЕТКА-ВЕТВИ]
    ( : [ОПИСАНИЯ-РЕЗУЛЬТАТОВ-ВЕТВИ] [✱ МЕТКА-ВЕТВИ] )+
ОПИСАНИЯ-РЕЗУЛЬТАТОВ-ВЕТВИ ::= ОПИСАНИЯ-РЕЗУЛЬТАТОВ
МЕТКА-ВЕТВИ ::= МЕТКА

```

После двоеточия описываются параметры-результаты очередной ветви гиперфункции. Ветви гиперфункции именуется метками. Если в описании результатов ветви нет метки, то ветвь именуется целым – порядковым номером ветви начиная с 1. Исполнение гиперфункции завершается выбором одной из ветвей и вычислением значений результатов этой ветви. При этом результаты других ветвей не определены. Оператор перехода **#имяВетви** в теле предиката завершает исполнение гиперфункции ветвью с именем **имяВетви**.

```

ПРЕДУСЛОВИЯ-ГИПЕРФУНКЦИИ ::= [ОБЩЕЕ-ПРЕДУСЛОВИЕ] (ПРЕДУСЛОВИЕ-ВЕТВИ)+
ОБЩЕЕ-ПРЕДУСЛОВИЕ ::= pre ПРЕДУСЛОВИЕ
ПРЕДУСЛОВИЕ-ВЕТВИ ::= pre МЕТКА-ВЕТВИ : ПРЕДУСЛОВИЕ

```

Отсутствие общего предусловия определяет допустимость произвольных значений аргументов в соответствии с их типами. Предусловие ветви определяет дополнительное условие на значения аргументов, при котором исполнение гиперфункции завершается выбором этой ветви. При этом значения аргументов должны удовлетворять общему предусловию. Предусловие последней ветви обычно не указывается, поскольку оно получается дополнением предусловий предыдущих ветвей.

ПОСТУСЛОВИЯ-ВЕТВЕЙ ::= (ПОСТУСЛОВИЕ-ВЕТВИ)+  
 ПОСТУСЛОВИЕ-ВЕТВИ ::= **post** МЕТКА-ВЕТВИ : ПОСТУСЛОВИЕ

Постусловие ветви связывает значения аргументов и результатов ветви. Ветвь без результатов постусловия не имеет.

### 3.2. Спецификация предиката

Предикат, используемый в программе, должен иметь определение предиката. Если предикат определяется в другом модуле программы, то предикат может быть представлен своей спецификацией.

СПЕЦИФИКАЦИЯ-ПРЕДИКАТА ::= ИМЯ-ПРЕДИКАТА ОПИСАНИЕ-СПЕЦИФИКАЦИИ-ПРЕДИКАТА

Имя предиката обозначает предикат, представленный описанием спецификации предиката.

ОПИСАНИЕ-СПЕЦИФИКАЦИИ-ПРЕДИКАТА ::=  
 ОПИСАНИЕ-СПЕЦИФИКАЦИИ-ПРЕДИКАТА-ФУНКЦИИ |  
 ОПИСАНИЕ-СПЕЦИФИКАЦИИ-ПРЕДИКАТА-ГИПЕРФУНКЦИИ  
 ОПИСАНИЕ-СПЕЦИФИКАЦИИ-ПРЕДИКАТА-ФУНКЦИИ ::=  
 ЗАГОЛОВОК-ПРЕДИКАТА [**pre** ПРЕДУСЛОВИЕ] [**post** ПОСТУСЛОВИЕ]  
 ОПИСАНИЕ-СПЕЦИФИКАЦИИ-ПРЕДИКАТА-ГИПЕРФУНКЦИИ ::=  
 ЗАГОЛОВОК-ПРЕДИКАТА-ГИПЕРФУНКЦИИ [ПРЕДУСЛОВИЯ-ГИПЕРФУНКЦИИ]  
 [ПОСТУСЛОВИЯ-ВЕТВЕЙ]

### 3.3. Вызов предиката

Элементарным оператором программы является вызов предиката.

ВЫЗОВ-ПРЕДИКАТА ::=  
 ВЫЗОВ-ПРЕДИКАТА-ФУНКЦИИ | ВЫЗОВ-ПРЕДИКАТА-ГИПЕРФУНКЦИИ  
 ВЫЗОВ-ПРЕДИКАТА-ФУНКЦИИ ::=  
 ИДЕНТИФИКАЦИЯ-ПРЕДИКАТА ( [АРГУМЕНТЫ] : РЕЗУЛЬТАТЫ )  
 ИДЕНТИФИКАЦИЯ-ПРЕДИКАТА ::=  
 [ИМЯ-МОДУЛЯ .] ИМЯ-ПРЕДИКАТА | ВЫРАЖЕНИЕ

Выражение должно быть предикатного типа. Значением выражения является предикат, запускаемый на исполнение данным вызовом.

АРГУМЕНТЫ ::= СПИСОК-ВЫРАЖЕНИЙ  
 СПИСОК-ВЫРАЖЕНИЙ ::= ВЫРАЖЕНИЕ [, СПИСОК-ВЫРАЖЕНИЙ]

Типы аргументов вызова должны быть совместимы (см. разд. 7) с типами соответствующих аргументов определения (или спецификации) вызываемого предиката.

```

РЕЗУЛЬТАТЫ ::= ПЕРЕМЕННАЯ [, РЕЗУЛЬТАТЫ] |
                РЕЗУЛЬТАТ-ЛОКАЛ [, РЕЗУЛЬТАТЫ]
РЕЗУЛЬТАТ-ЛОКАЛ ::= ОПИСАНИЕ-ПЕРЕМЕННОЙ
ОПИСАНИЕ-ПЕРЕМЕННОЙ ::= ИЗОБРАЖЕНИЕ-ТИПА-ПЕРЕМЕННОЙ ИДЕНТИФИКАТОР
ИЗОБРАЖЕНИЕ-ТИПА-ПЕРЕМЕННОЙ ::=
                ИЗОБРАЖЕНИЕ-ТИПА [:blank:] | var [:blank:]

```

Использование `var` возможно вместо соответствующего ИЗОБРАЖЕНИЕ-ТИПА, поскольку тип результата-локала в большинстве случаев можно восстановить по определению вызываемого предиката. Описатель `var` также может использоваться при описании локальной переменной в случае, когда тип этой переменной легко определяется из контекста дальнейшего использования переменной (см. разд. 4, 5).

Типы результатов вызова должны совпадать с типами соответствующих результатов в определении вызываемого предиката. Переменная, представленная описанием результата-локала, определяется как локальная в теле предиката, содержащем данный вызов. Описание локальной результирующей переменной внутри вызова эквивалентно описанию этой переменной перед вызовом.

Использование специального имени “`_`” в качестве результирующей переменной обозначает пустой результат: результат по этой позиции, получаемый при завершении исполнения вызова предиката, не используется в дальнейшем исполнении. Имя “`_`” зарезервировано в языке P – его нельзя использовать для именованя объектов программы.

```

ВЫЗОВ-ПРЕДИКАТА-ГИПЕРФУНКЦИИ ::=
                ИДЕНТИФИКАЦИЯ-ПРЕДИКАТА ( [АРГУМЕНТЫ] (: РЕЗУЛЬТАТЫ-ВЕТВИ)+ )
РЕЗУЛЬТАТЫ-ВЕТВИ ::= [РЕЗУЛЬТАТЫ] [ОПЕРАТОР-ПЕРЕХОДА]
ОПЕРАТОР-ПЕРЕХОДА ::= * МЕТКА

```

В качестве метки в операторе указывается либо метка ветви предиката, в теле которого находится данный вызов, либо метка ветви обработчика; подробнее см. в разд. 5.

Исполнение вызова предиката реализуется следующим образом. Результатом исполнения аргументов вызова является набор значений. Вычисление каждого аргумента вызова реализуется независимо от вычисления других, т.е. параллельно. Полученный набор значений аргументов присваивается соответствующим входным параметрам определения вызываемого предиката. Далее исполняется тело определения вызываемого предиката. В процессе исполнения тела определяется итоговая ветвь предиката и вычисляются значения результирующих переменных по этой ветви. Наконец, полученные значения результирующих пе-

ременных присваиваются соответствующим результирующим переменным вызова предиката, и исполнение вызова завершается по этой ветви вызова. Если в данной ветви вызова указан оператор перехода, то либо происходит переход на локальную метку, либо завершается тело предиката, содержащего вызов, той ветвью, которая указана в операторе перехода. Отметим, что подстановка аргументов и результатов предиката реализуется по значению.

**ВЫЗОВ-ФУНКЦИИ ::= ИДЕНТИФИКАЦИЯ-ПРЕДИКАТА ( [АРГУМЕНТЫ] )**

Вызов предиката-функции может иметь вид вызова функции. Результатом исполнения является набор значений результирующих переменных исполненного определения предиката.

```

1  real r;
2  sqrt (2 : r);
3  real q = sqrt (3);
4  // спецификация гиперфункции Comp:
5  Comp (list(int) s: int d, list (int) r)
6  pre  1: s = nil
7  post 2: s = cons(d, r) ;
8  /* если список s пуст, реализуется первая ветвь
9     гиперфункции, иначе – реализуется вторая ветвь
10     с результатами: d – первый элемент, r – хвост списка */
11
12 elemTwo (list(int) s: int e #value : #empty)
13 pre  empty: s = nil or s.cdr = nil
14 {
15     Comp (s : #empty : int e1, list(int) s1 #ok)
16         case ok: Comp (s1 : #empty : e, list(int) s2 #value);
17 }
18 post value: e = (s.cdr).car;
19 /* гиперфункция elemTwo извлекает второй элемент списка,
20    если элемент существует */

```

**Пример 2. Вызовы и определения предикатов и гиперфункций, спецификация гиперфункции**

В приведенном выше определении предиката `elemTwo` переход по локальной метке `ok` является излишним. Кроме того, результаты `e1` и `s2` далее нигде не используются. Определение предиката `elemTwo`, представленное ниже, исправляет отмеченные недостатки:

```

1 elemTwo (list(int) s: int e #value : #empty)
2 pre 1: s = nil or s.cdr = nil
3 {      Comp (s : #empty : _, list(int) s1 );
4      Comp (s1 : #empty : e, _ #value);
5 }
6 post value: e = s.cdr.car;

```

## 4. ПРОГРАММА

Программа состоит из одного или нескольких модулей. Модуль определяет независимую часть программы.

ОПИСАНИЕ-МОДУЛЯ ::= [ЗАГОЛОВОК-МОДУЛЯ] СПИСОК-ОПИСАНИЙ  
 ЗАГОЛОВОК-МОДУЛЯ ::= **module** ИМЯ-МОДУЛЯ [ ( ОПИСАНИЯ-АРГУМЕНТОВ ) ];  
 ИМЯ-МОДУЛЯ ::= ИДЕНТИФИКАТОР

Аргументы, описанные в круглых скобках, являются формальными параметрами модуля. Структура ОПИСАНИЙ-АРГУМЕНТОВ определена в разд. 3.1.

СПИСОК-ОПИСАНИЙ ::= ОПИСАНИЕ ( ; ОПИСАНИЕ ) \*  
 ОПИСАНИЕ ::=

ИМПОРТ-МОДУЛЯ | ОПИСАНИЕ-ТИПА | ОПИСАНИЕ-ГЛОБАЛЬНЫХ-ПЕРЕМЕННЫХ |  
 ОПРЕДЕЛЕНИЕ-ПРЕДИКАТА | СПЕЦИФИКАЦИЯ-ПРЕДИКАТА |  
 ОПИСАНИЕ-ФОРМУЛЫ | ОПРЕДЕЛЕНИЕ-КЛАССА | ОПИСАНИЕ-СООБЩЕНИЯ |  
 ОПРЕДЕЛЕНИЕ-ПРОЦЕССА

ОПИСАНИЕ-ФОРМУЛЫ (см. разд. 9) определяет функцию или предикат; они встречаются в предусловиях, постуловиях и формулах. Три последние альтернативы ОПИСАНИЯ используются для задания процессов; см. разд. 10.

ИМПОРТ-МОДУЛЯ ::=  
**import** ИМЯ-МОДУЛЯ [ ( АРГУМЕНТЫ ) ] [ **as** ЛОКАЛЬНОЕ-ИМЯ-МОДУЛЯ ]  
 ЛОКАЛЬНОЕ-ИМЯ-МОДУЛЯ ::= ИДЕНТИФИКАТОР

Имена модулей, встречающиеся в описаниях данного модуля, должны быть определены конструкцией ИМПОРТ-МОДУЛЯ. Если импортируемый модуль имеет формальные параметры, то в круглых скобках задается соответствующий набор фактических параметров. Конструкция АРГУМЕНТЫ определена в разд. 3.3. Правила подстановки фактических параметров на место формальных те же самые, что и для вызова предиката. ЛОКАЛЬНОЕ-ИМЯ-МОДУЛЯ используется в теле модуля как эквивалент ИМЯ-МОДУЛЯ ( АРГУМЕНТЫ ).

ОПИСАНИЕ-ГЛОБАЛЬНЫХ-ПЕРЕМЕННЫХ ::= ОПИСАНИЕ-ПЕРЕМЕННЫХ |  
ОПИСАНИЕ-ПЕРЕМЕННЫХ-С-ИНИЦИАЛИЗАЦИЕЙ

Глобальные переменные обычно определяют внешние параметры задачи. В нижеследующих определениях предикатов эти переменные считаются аргументами. Они используются в теле предиката, но не упоминаются в заголовке.

ОПИСАНИЕ-ПЕРЕМЕННЫХ ::=  
ИЗОБРАЖЕНИЕ-ТИПА-ПЕРЕМЕННОЙ ИМЯ-ПЕРЕМЕННОЙ (, ИМЯ-ПЕРЕМЕННОЙ)\*  
ОПИСАНИЕ-ПЕРЕМЕННЫХ-С-ИНИЦИАЛИЗАЦИЕЙ ::=  
ИЗОБРАЖЕНИЕ-ТИПА-ПЕРЕМЕННОЙ  
ОПРЕДЕЛЕНИЕ-ПЕРЕМЕННОЙ (, ОПРЕДЕЛЕНИЕ-ПЕРЕМЕННОЙ)\*  
ОПРЕДЕЛЕНИЕ-ПЕРЕМЕННОЙ ::= ИМЯ-ПЕРЕМЕННОЙ = ВЫРАЖЕНИЕ |  
ИМЯ-ПЕРЕМЕННОЙ

Тип выражения должен быть совместим с типом переменной, которой присваивается значение переменной.

Модуль определяет совокупность имен и обозначаемых ими объектов: предикатов, типов, переменных, полей структур, конструкторов и распознавателей объединений, процессов, классов и сообщений. Объекты, представленные описаниями модуля, определяют область глобальных имен. Каждое описание предиката определяет автономную систему локализации имен: параметров, меток и локальных переменных. В описании предиката одно имя не может использоваться для двух разных объектов. Если глобальное имя используется в определении предиката, то это имя не может определяться в качестве имени метки, локальной переменной или параметра.

Следующей областью локализации являются конструкции: изображение подтипа, определение массива, элемент агрегата вида “итерация выражений”. Переменные, определенные в каждой из этих конструкций, локализованы в конструкции и недоступны вне нее. Класс также определяет независимую область локализации. Поля класса доступны префиксацией объекта класса (см. разд. 6).

Исполнение программы начинается исполнением предиката с именем `main`. Рекомендован предикат со следующими параметрами: `main (list(string) argv : int ret_code)`. Параметр `argv` предоставляет массив входных аргументов при исполнении программы. Параметру `ret_code` присваивается значение кода выхода.

Для вывода информации используется оператор `print`, аргументы которого – выражения и массивы.

## 5. ОПЕРАТОРЫ

БЛОК ::= { ОПЕРАТОР }  
ОПЕРАТОР ::= [ОПИСАНИЕ-ПЕРЕМЕННЫХ ;]{ ЗАМКНУТЫЙ-ОПЕРАТОР |  
[ОПИСАНИЕ-ПЕРЕМЕННЫХ ;] ПАРАЛЛЕЛЬНЫЙ-ОПЕРАТОР |  
[ОПИСАНИЕ-ПЕРЕМЕННЫХ ;] ОПЕРАТОР-СУПЕРПОЗИЦИИ

Описанные переменные считаются локальными в теле предиката. Тип локальной переменной может быть также определен описателем `var` (см. разд. 3.3) в случае, когда тип легко определяется из контекста дальнейшего использования переменной.

ЗАМКНУТЫЙ-ОПЕРАТОР ::=  
ОПЕРАТОР-ПРИСВАИВАНИЯ | ВЫЗОВ-ПРЕДИКАТА-ФУНКЦИИ |  
ВЫЗОВ-ПРЕДИКАТА-ГИПЕРФУНКЦИИ [ОПЕРАТОР-ОБРАБОТКИ-ВЕТВЕЙ] |  
УСЛОВНЫЙ-ОПЕРАТОР | БЛОК [ОПЕРАТОР-ОБРАБОТКИ-ВЕТВЕЙ] |  
ОПЕРАТОР-ВЫБОРА | ОПРЕДЕЛЕНИЕ-ЛОКАЛЬНОГО-ПРЕДИКАТА  
ОПЕРАТОР-ПРИСВАИВАНИЯ ::= ПЕРЕМЕННАЯ = ВЫРАЖЕНИЕ

В результате исполнения оператора присваивания переменной присваивается значение выражения. Типы выражения и переменной должны быть совместимы (см. разд. 7).

УСЛОВНЫЙ-ОПЕРАТОР ::=  
ОПЕРАТОР-ЕСЛИ **else** ЗАМКНУТЫЙ-ОПЕРАТОР [ОПЕРАТОР-ПЕРЕХОДА]  
ОПЕРАТОР-ЕСЛИ ::=  
**if** ( ВЫРАЖЕНИЕ ) ЗАМКНУТЫЙ-ОПЕРАТОР [ОПЕРАТОР-ПЕРЕХОДА]

В операторе перехода указывается либо метка ветви гиперфункции, либо метка ветви обработчика, следующего за блоком, содержащим данный условный оператор.

```
1 if ( x > 0 ) s = 1
2 else if ( x < 0 ) s = -1
3     else s = 0
```

Пример 3. Использование условного оператора

ПАРАЛЛЕЛЬНЫЙ-ОПЕРАТОР ::= ЗАМКНУТЫЙ-ОПЕРАТОР (|| ЗАМКНУТЫЙ-ОПЕРАТОР)+

Операторы, образующие параллельный оператор, должны иметь непересекающиеся наборы результирующих переменных. Исполнение параллельного оператора складывается из исполнения входящих в него операторов. Операторы выполняются независимо друг от друга; они могут исполняться параллельно.

ОПЕРАТОР-СУПЕРПОЗИЦИИ ::= ОПЕРАТОР ( ; ОПЕРАТОР)+

В цепочке операторов, составляющих оператор суперпозиции, каждый следующий оператор использует значения локальных переменных, присвоенных предыдущими операторами. Каждая пара соседних операторов в цепочке должна быть связана хотя бы одной локальной переменной; если такой связи нет, их композиция должна быть оформлена параллельным оператором.

```

ЗАГОЛОВОК-ОПЕРАТОРА-ВЫБОРА ::= switch ( ВЫРАЖЕНИЕ )
ОПЕРАТОР-ВЫБОРА             ::= ЗАГОЛОВОК-ОПЕРАТОРА-ВЫБОРА
                               { ОПЕРАТОР-АЛЬТЕРНАТИВЫ+
                               [ default : ОПЕРАТОР [ОПЕРАТОР-ПЕРЕХОДА] ]
                               }
ОПЕРАТОР-АЛЬТЕРНАТИВЫ ::= case АЛЬТЕРНАТИВА ( , АЛЬТЕРНАТИВА)* :
                               ОПЕРАТОР [ОПЕРАТОР-ПЕРЕХОДА]
АЛЬТЕРНАТИВА              ::= ВЫРАЖЕНИЕ | ОПРЕДЕЛЕНИЕ-КОНСТРУКТОРА

```

Выполняется тот оператор альтернативы, для которого значение АЛЬТЕРНАТИВЫ совпадает со значением выражения в заголовке оператора выбора. Оператор после **default** выполняется в случае, когда нет ОПЕРАТОРА-АЛЬТЕРНАТИВЫ с требуемым значением АЛЬТЕРНАТИВЫ. Траектория операторов перехода та же, что и для условного оператора.

Если в позиции выражения в заголовке оператора выбора находится переменная типа объединения, то в качестве АЛЬТЕРНАТИВЫ указывается ОПРЕДЕЛЕНИЕ-КОНСТРУКТОРА. Особенности выполнения ОПЕРАТОРА-АЛЬТЕРНАТИВЫ определены в разд. 7.

```

ОПЕРАТОР-ОБРАБОТКИ-ВЕТВЕЙ ::=
                               ( case [МЕТКА-ВЕТВИ-ОБРАБОТЧИКА :] ОПЕРАТОР)+
МЕТКА-ВЕТВИ-ОБРАБОТЧИКА ::= МЕТКА

```

Оператор обработки ветвей следует за вызовом гиперфункции либо за блоком, содержащим операторы перехода. Если предыдущий оператор нормально завершается, то оператор обработки ветвей игнорируется. Нормальное завершение возможно и для вызова гиперфункции, когда на одной из ветвей гиперфункции опущена метка перехода, что означает суперпозицию (или параллельную композицию) с последующим оператором, а на другой ветви имеется переход на метку ветви обработчика или на метку ветви гиперфункции, в теле которой находится данный оператор.

Исполнение оператора обработки ветвей инициируется оператором перехода, находящимся в предыдущем операторе. При этом исполняется оператор, помеченный соответствующей меткой, после чего исполнение оператора обработки ветвей и предыдущего оператора считается

завершившимся.

В вызове гиперфункции могут быть опущены переходы, а в последующем операторе обработки ветвей – метки ветвей обработчика. В этом случае порядок ветвей обработчика соответствует порядку ветвей вызова гиперфункции.

```
1 A(i: y #1: #2)
2   case 1: B(y, i: u)
3   case 2: D(i: u);
```

Пример 4. Вызов гиперфункции с обработчиком ветвей

**ОПРЕДЕЛЕНИЕ-ЛОКАЛЬНОГО-ПРЕДИКАТА ::= ОПРЕДЕЛЕНИЕ-ПРЕДИКАТА**

Определение локального предиката допускает использование параметров и локальных переменных предиката, в теле которого это определение находится. Локальный предикат доступен лишь в теле предиката, внутри которого он описан.

## 6. ВЫРАЖЕНИЯ

**ПЕРВИЧНОЕ-ВЫРАЖЕНИЕ ::=**

КОНСТАНТА | ПЕРЕМЕННАЯ | АГРЕГАТ | ИМЯ-ПРЕДИКАТА |  
ГЕНЕРАТОР-ПРЕДИКАТА | ВЫЗОВ-ФУНКЦИИ | ( ВЫРАЖЕНИЕ ) | ИМЯ-ТИПА |  
МОДИФИКАЦИЯ | ЭЛЕМЕНТ-СПИСКА | ОПРЕДЕЛЕНИЕ-МАССИВА |  
КОНСТРУКТОР | РАСПОЗНАВАТЕЛЬ | ПОЛЕ-ОБЪЕДИНЕНИЯ

Имя типа может быть аргументом вызова предиката и изображения типа. Определение ЭЛЕМЕНТА-СПИСКА дается в разд. 7. Операция ОПРЕДЕЛЕНИЕ-МАССИВА описывается в разд. 8.3. Последние три альтернативы в определении ПЕРВИЧНОГО-ВЫРАЖЕНИЯ относятся к объектам типа объединения и описаны в разд. 7.

**ПЕРЕМЕННАЯ ::= ИМЯ-ПЕРЕМЕННОЙ | ЭЛЕМЕНТ-МАССИВА | ПОЛЕ-СТРУКТУРЫ |**  
**ПЕРЕМЕННАЯ-КЛАССА | ВЫРЕЗКА-МАССИВА | МУЛЬТИПЕРЕМЕННАЯ**

**ВЫРЕЗКА-МАССИВА** определена в разд. 8.2.

**ИМЯ-ПЕРЕМЕННОЙ ::= ИДЕНТИФИКАТОР [']**

Переменная вида *имя'* используется для именованного результата предиката. Переменная *имя'* склеивается с переменной *имя*, являющейся аргументом предиката (см. разд. 3.1).

**ЭЛЕМЕНТ-МАССИВА ::= ПЕРВИЧНОЕ-ВЫРАЖЕНИЕ [ ИНДЕКСЫ-МАССИВА ]**  
**ИНДЕКСЫ-МАССИВА ::= СПИСОК-ВЫРАЖЕНИЙ**

Значением ПЕРВИЧНОГО-ВЫРАЖЕНИЯ является переменная типа “массив”. Значения ИНДЕКСОВ-МАССИВА должны принадлежать типам индексов соответствующего типа массива.

ПОЛЕ-СТРУКТУРЫ ::= ПЕРВИЧНОЕ-ВЫРАЖЕНИЕ . ИМЯ-ПОЛЯ

Значением ПЕРВИЧНОГО-ВЫРАЖЕНИЯ является переменная типа “структура”.

ИМЯ-ПОЛЯ ::= ИДЕНТИФИКАТОР

ПЕРЕМЕННАЯ-КЛАССА ::= ПЕРВИЧНОЕ-ВЫРАЖЕНИЕ . ПЕРЕМЕННАЯ

Значением ПЕРВИЧНОГО-ВЫРАЖЕНИЯ является переменная типа “класс”.

МУЛЬТИПЕРЕМЕННАЯ ::= | СПИСОК-ПЕРЕМЕННЫХ | |  
СПИСОК-ПЕРЕМЕННЫХ

Два разных способа представления мультипеременной считаются равноценными.

СПИСОК-ПЕРЕМЕННЫХ ::= ПЕРЕМЕННАЯ [, СПИСОК-ПЕРЕМЕННЫХ ]

Мультипеременная определяет упорядоченный набор переменных. Число переменных в наборе должно быть больше 1.

АГРЕГАТ ::= АГРЕГАТ-СТРУКТУРА | АГРЕГАТ-МАССИВ | АГРЕГАТ-МНОЖЕСТВО |  
АГРЕГАТ-СПИСОК

Агрегат определяет значение структурного типа. Агрегат-массив определяет массив, агрегат-множество – подмножество некоторого множества – значение типа **set** (см. разд. 7), агрегат-структура – структуру (как набор полей), агрегат-список – список, заданный последовательностью элементов.

АГРЕГАТ-СТРУКТУРА ::= [ИЗОБРАЖЕНИЕ-ТИПА] ( ЭЛЕМЕНТЫ-АГРЕГАТА )

АГРЕГАТ-МАССИВ ::= [ИЗОБРАЖЕНИЕ-ТИПА] [ [ЭЛЕМЕНТЫ-АГРЕГАТА] ]

АГРЕГАТ-МНОЖЕСТВО ::= [ИЗОБРАЖЕНИЕ-ТИПА] { [ЭЛЕМЕНТЫ-АГРЕГАТА] }

АГРЕГАТ-СПИСОК ::= [ИЗОБРАЖЕНИЕ-ТИПА] [[ ЭЛЕМЕНТЫ-АГРЕГАТА ]]

Агрегат [ ] обозначает пустой массив (когда верхняя граница индексов меньше нижней). Агрегат { } определяет пустое множество. Если пара скобок [ ] или [ ] встречается рядом при изображении АГРЕГАТ-МАССИВА, то соседние скобки должны быть разделены пробелом.

ЭЛЕМЕНТЫ-АГРЕГАТА ::= ЭЛЕМЕНТ-АГРЕГАТА [, ЭЛЕМЕНТЫ-АГРЕГАТА]

Значением агрегата является набор значений элементов агрегата, перечисленных в скобках. Агрегат может иметь иерархическую структуру, поскольку элементом агрегата может быть агрегат. Тип агрегата определяется типом позиции, в которой находится агрегат. Значение

агрегата должно соответствовать типу позиции. Тип агрегата может быть указан явно ИЗОБРАЖЕНИЕМ-ТИПА – это не является необходимым, но может улучшить восприятие программы.

ЭЛЕМЕНТ-АГРЕГАТА ::= [ИНДЕКС-ЭЛЕМЕНТА-АГРЕГАТА :] ВЫРАЖЕНИЕ  
 ИНДЕКС-ЭЛЕМЕНТА-АГРЕГАТА ::= ВЫРАЖЕНИЕ | ИМЯ-ПОЛЯ

Индексы элементов, если таковые присутствуют, должны соответствовать индексам элементов массива или именам полей структуры.

```
1 type Ar4 = array (real, 1..4);
2 Ar4 x = [ 0, 1, 2, 3];
```

Пример 5. Задание агрегата

ГЕНЕРАТОР-ПРЕДИКАТА ::= **predicate** ОПИСАНИЕ-ПРЕДИКАТА

Исполнение генератора предиката создает новый предикат, операционная часть которого определяется телом предиката, находящегося в составе ОПИСАНИЯ-ПРЕДИКАТА; см. разд. 3.1. В теле предиката фиксируются значения свободных переменных (отличных от аргументов и результатов) на момент исполнения генератора. Новый предикат становится значением генератора предиката.

УНАРНОЕ-ВЫРАЖЕНИЕ ::= УНАРНАЯ-ОПЕРАЦИЯ ПЕРВИЧНОЕ-ВЫРАЖЕНИЕ  
 УНАРНАЯ-ОПЕРАЦИЯ ::= + | - | ! | ^  
 БИНАРНОЕ-ВЫРАЖЕНИЕ ::= ВЫРАЖЕНИЕ БИНАРНАЯ-ОПЕРАЦИЯ ВЫРАЖЕНИЕ  
 БИНАРНАЯ-ОПЕРАЦИЯ ::= \* | / | % | + | - | << | >> | in | < | > | <= | >= | = | != | & | ^ | or | xor | => | <=>

Семантика операций и их приоритеты определены ниже в таблице. Типы аргументов каждой операции должны соответствовать операции. Операция “+” для объединения массивов описана в разд. 8.4.

МОДИФИКАЦИЯ ::= МОДИФИКАЦИЯ-СТРУКТУРЫ | МОДИФИКАЦИЯ-МАССИВА  
 МОДИФИКАЦИЯ-СТРУКТУРЫ ::= ВЫРАЖЕНИЕ **with** АГРЕГАТ  
 МОДИФИКАЦИЯ-МАССИВА ::= ВЫРАЖЕНИЕ **with** АГРЕГАТ |  
 ВЫРАЖЕНИЕ **with** ОПРЕДЕЛЕНИЕ-МАССИВА-ПО-ЧАСТЯМ

В структурном значении выражения, обозначающего массив или структуру, меняются компоненты, представленные АГРЕГАТОМ или ОПРЕДЕЛЕНИЕМ-МАССИВА-ПО-ЧАСТЯМ (см. разд. 8.3). Значением операции является обновленный массив или структура.

## Операции в порядке уменьшения приоритета

$\wedge$	Возведение в степень
$+$ , $-$ , $!$ , $\sim$	Унарные плюс и минус, логическое отрицание и поэлементное дополнение для множеств, побитовое дополнение для ограниченных целых
$*$ , $/$ , $\%$	Арифметическое умножение, деление и остаток от целочисленного деления
$+$ , $-$	Арифметическое сложение и вычитание; объединение и вычитание множеств; конкатенация списков; объединение массивов с непересекающимися типами индексов
$\ll$ , $\gg$	Побитовый (поэлементный) сдвиг влево и вправо для целых
<b>in</b>	Проверка вхождения элемента в множество
$<$ , $>$ , $<=$ , $>=$	Операции арифметического сравнения
$=$ , $!=$	Проверка на равенство и неравенство
<b>&amp;</b>	Пересечение множеств, логическая конъюнкция, побитовое И для ограниченных целых
<b>xor</b>	Симметрическая разность для множеств, исключительное ИЛИ для логических выражений и ограниченных целых (побитовое)
<b>or</b>	Объединение множеств, логическая дизъюнкция, побитовое ИЛИ для ограниченных целых
$=>$	Импликация
$<=>$	Логическое тождество
<b>?:</b>	Трёхместная условная операция

```

1   type ar0_4 = array (int , 0..4);
2   ar0_4 ones = [ 1, 1, 1, 2, 1 ];
3   ar0_4 ones1 = ones [3: 1]; // [ 1, 1, 1, 1, 1 ]
4   type Vec(nat n) = array (real , 1..n);
5   perm(nat n, Vec(n) b, nat m, k : Vec(n) b')
6   { b' = b with [k: b [m], m: b [k]] };

```

Пример 6. Замена части структурного значения

УСЛОВНОЕ-ВЫРАЖЕНИЕ ::= ВЫРАЖЕНИЕ ? ВЫРАЖЕНИЕ : ВЫРАЖЕНИЕ

Первое выражение имеет тип “логический”. Если первое ВЫРАЖЕНИЕ завершается идентификатором, то идентификатор и последующий символ “?” должны быть разделены пробелом.

МУЛЬТИВЫРАЖЕНИЕ ::= | СПИСОК-ВЫРАЖЕНИЙ |  
СПИСОК-ВЫРАЖЕНИЙ

Значением мультिवыражения является набор значений, получающихся в результате вычисления перечисленных выражений. Выражения вычисляются независимо, т.е. параллельно. Мультिवыражение может находиться в правой части оператора присваивания, если в левой находится мультипеременная.

ВЫРАЖЕНИЕ ::= ПЕРВИЧНОЕ-ВЫРАЖЕНИЕ | УНАРНОЕ-ВЫРАЖЕНИЕ |  
БИНАРНОЕ-ВЫРАЖЕНИЕ | УСЛОВНОЕ-ВЫРАЖЕНИЕ |  
МУЛЬТИВЫРАЖЕНИЕ

## 7. ТИПЫ

Всякая переменная имеет некоторый тип. Тип может быть атрибутом языковой конструкции. Семантика конструкции налагает определенные ограничения на значения типов. Для каждой подконструкции некоторой конструкции определяется *позиция* ее вхождения внутри конструкции. Это, например, позиции операндов операций, позиции правой и левой частей в операторе присваивания, позиция фактического параметра в вызове предиката и др. При наличии в конструкции двух позиций типы этих позиций должны быть *согласованы*. Частным случаем согласованности является *совместимость* типов. Например, тип выражения правой части оператора присваивания должен быть совместим с типом переменной левой части оператора присваивания, но не наоборот. Реализуется неявное преобразование (называемое *приведением*) значения совместимого типа к типу, требуемому позицией.

ОПИСАНИЕ-ТИПА ::=  
**type** ИМЯ-ТИПА [( ПАРАМЕТРЫ-ТИПА )] = ИЗОБРАЖЕНИЕ-ТИПА |  
ПРЕДОПИСАНИЕ-ТИПА  
ИМЯ-ТИПА ::= ИДЕНТИФИКАТОР  
ПАРАМЕТРЫ-ТИПА ::=  
ИЗОБРАЖЕНИЕ-ТИПА [:blank:] ИДЕНТИФИКАТОР (, ИДЕНТИФИКАТОР)\*  
[, ПАРАМЕТРЫ-ТИПА]

Описание типа связывает имя типа с его изображением. Тип может быть параметризован, т.е. зависеть от набора значений переменных и типов, указанных в качестве параметров.

**ПРЕДОПИСАНИЕ-ТИПА ::= type ИМЯ-ТИПА**

Если имя типа используется до его описания, что возможно для рекурсивных определений типов, то перед первым вхождением имени оно должно быть декларировано с помощью предописания.

**ИЗОБРАЖЕНИЕ-ТИПА ::=**

ИЗОБРАЖЕНИЕ-ПРИМИТИВНОГО-ТИПА | ИЗОБРАЖЕНИЕ-ТИПА-ПО-ИМЕНИ |  
ИЗОБРАЖЕНИЕ-ТИПА-КАК-ПАРАМЕТРА | ИЗОБРАЖЕНИЕ-ТИПА-ПРЕДИКАТА |  
ИЗОБРАЖЕНИЕ-ПОДТИПА | ИЗОБРАЖЕНИЕ-ПЕРЕЧИСЛЕНИЯ |  
ИЗОБРАЖЕНИЕ-СТРУКТУРНОГО-ТИПА | ИЗОБРАЖЕНИЕ-ТИПА-ОБЪЕДИНЕНИЯ

**ИЗОБРАЖЕНИЕ-ПРИМИТИВНОГО-ТИПА ::=**

**nat** [(РАЗРЯДНОСТЬ-ЦЕЛЫХ)]  
| **int** [(РАЗРЯДНОСТЬ-ЦЕЛЫХ)]  
| **real** [(РАЗРЯДНОСТЬ-ВЕЩЕСТВЕННЫХ)]  
| **bool**  
| **char**

Тип **nat** является подмножеством неотрицательных чисел типа **int**, **char** – множество символов некоторого алфавита. Конкретные представления примитивных типов даются в императивном расширении языка (см. разд. 11).

Разрядность определяет максимальное число битов в представлении значений типа. Указание разрядности в изображении примитивного типа является частью императивного расширения языка.

Значения типа **nat** совместимы с типом **int**, а **int** - с типом **real**. Значения типа меньшей размерности совместимы с аналогичными типами большей размерности.

**ИЗОБРАЖЕНИЕ-ТИПА-ПО-ИМЕНИ ::=**

[ИМЯ-МОДУЛЯ .] ИМЯ-ТИПА [( АРГУМЕНТЫ )]

Ранее определенный тип может быть представлен своим именем. Изображение параметризованного типа представляется с конкретными параметрами.

**ИЗОБРАЖЕНИЕ-ТИПА-КАК-ПАРАМЕТРА ::= type ИМЯ-ТИПА**

Объявление типа в качестве параметра описания типа или определения предиката реализуется описателем **type**. В качестве обозначения типа может использоваться выражение типа **type**, которым, например, может являться параметр предиката или параметризованного типа. Например:

```

1 type Point (type T) = struct (T x, y);
2 type Polyline (type T) = list (Point (T));
3
4 decimate (type T, list (T) data : list (T) data') {
5   data' = (data = nil) ?
6     nil :
7     data.car +
8     (len(data) < 3 ? nil : decimate(T, data.cdr.cdr))
9 };
10
11 reverse (type T, list (T) data : list (T) data') {
12   data' = (data = nil) ? nil :
13     reverse (T, data.cdr) + data.car
14 };
15
16 Polyline (real) line = [[ (0.0, 0.0), (0.1, 0.2),
17   (0.2, 0.5), (0.3, 0.6),
18   (0.4, 0.3), (0.5, 0.0) ]];
19
20 Polyline (real) line1 = reverse (decimate (real, line));
21 // [[ (0.4, 0.3), (0.2, 0.5), (0.0, 0.0) ]]

```

#### Пример 7. Использование типов в качестве параметров

В примере выше идентификатор `T` использован в качестве типа-параметра, а тип `real` – как фактический параметр.

```

1 squareVec (type T, list (T) data : list (T) data') {
2   data' = (data = nil) ?
3     nil : data.car ^ 2 + squareVec (T, data.cdr)
4 };
5
6 list (int) squares = squareVec (int, [[ 0, 1, 2, 3, 4 ]]);
7 // = [[ 0, 1, 4, 9, 16 ]]

```

Выше приведен еще один пример использования типа `T` в качестве параметра предиката.

**ИЗОБРАЖЕНИЕ-ТИПА-ПРЕДИКАТА ::=**

**predicate** ОПИСАНИЕ-СПЕЦИФИКАЦИИ-ПРЕДИКАТА

Имя параметра предиката может быть опущено. В этом случае описывается только тип параметра. Изображение типа предиката используется для описания переменной предикатного типа.

```

1 type int_list = list (int);
2 map_list (int_list src, predicate (int : int)
3         func : int_list dest) {
4     dest = (src = nil) ?
5         nil : func (src.car) + map_list (src.cdr, func))
6 };
7 int_list numbers = [[ 0, 1, 2, 3, 4, 5 ]];
8 int_list squares =
9     map_list (numbers, predicate (int a : int b)
10            { b = a * a; });
11 // squares = [[ 0, 1, 4, 9, 16, 25 ]]

```

Пример 8. Использование предикатного типа

**ИЗОБРАЖЕНИЕ-ПОДТИПА ::=**

```

subtype ( ОПИСАНИЕ-ПЕРЕМЕННОЙ : ВЫРАЖЕНИЕ ) |
ИЗОБРАЖЕНИЕ-ДИАПАЗОНА

```

Конструкция **subtype**(T x: выражение) определяет подмножество типа T. Описание переменной x действует в пределах этой конструкции. Множество значений переменной x, для которых логическое **ВЫРАЖЕНИЕ** имеет значение “истина”, является определяемым подтипом типа T. Если **ВЫРАЖЕНИЕ** содержит другие переменные, то все они являются параметрами изображаемого типа. Конструкция **ОПИСАНИЕ-ПЕРЕМЕННОЙ** определена в разд. 3.3.

**ИЗОБРАЖЕНИЕ-ДИАПАЗОНА ::= ВЫРАЖЕНИЕ .. ВЫРАЖЕНИЕ**

Изображение диапазона является подмножеством типа **int**, **enum** или **char**. Изображение типа **subtype(int i: i>=1 & i<=n+1)** эквивалентно изображению диапазона: **1..n+1**.

```

1 type Odd = subtype (int n: n % 2 = 1);
2 type diap10_20 = 10..20;

```

**ИЗОБРАЖЕНИЕ-СТРУКТУРНОГО-ТИПА ::=**

```

ИЗОБРАЖЕНИЕ-ТИПА-СТРУКТУРЫ | ИЗОБРАЖЕНИЕ-ТИПА-ОБЪЕДИНЕНИЯ |
ИЗОБРАЖЕНИЕ-ТИПА-СПИСКА | ИЗОБРАЖЕНИЕ-ТИПА-МНОЖЕСТВА |
ИЗОБРАЖЕНИЕ-ТИПА-МАССИВА

```

**ИЗОБРАЖЕНИЕ-ТИПА-СТРУКТУРЫ ::= struct** ( ОПИСАНИЕ-ПОЛЕЙ )

```

ОПИСАНИЕ-ПОЛЕЙ      ::= ИЗОБРАЖЕНИЕ-ТИПА СПИСОК-ИМЕН-ПОЛЕЙ
                    [, ОПИСАНИЕ-ПОЛЕЙ]

```

```

СПИСОК-ИМЕН-ПОЛЕЙ  ::= ИМЯ-ПОЛЯ [, СПИСОК-ИМЕН-ПОЛЕЙ]

```

Значение типа структуры состоит из совокупности значений полей структуры. Имена полей должны быть различны. Число полей должно быть не менее двух.

```
1 type Point = struct ( int x, y );
2 Point pt = (10, 20);
```

### Пример 9. Определение структуры

ИЗОБРАЖЕНИЕ-ТИПА-ОБЪЕДИНЕНИЯ ::= **union** ( ОПИСАНИЯ-КОНСТРУКТОРОВ )  
ОПИСАНИЯ-КОНСТРУКТОРОВ ::= ОПИСАНИЕ-КОНСТРУКТОРА  
(, ОПИСАНИЕ-КОНСТРУКТОРА)\*

Значением типа объединения является значение одного из конструкторов, перечисленных в списке описаний конструкторов. Число конструкторов должно быть не меньше двух.

ОПИСАНИЕ-КОНСТРУКТОРА ::= ИМЯ-КОНСТРУКТОРА [ ( ОПИСАНИЕ-ПОЛЕЙ ) ]  
ИМЯ-КОНСТРУКТОРА ::= ИДЕНТИФИКАТОР

Поля, определяемые в круглых скобках, являются полями объединения. Значения этих полей используются в качестве аргументов конструктора. Имена полей объединения по всем конструкторам должны быть различны.

Работа с объектами типа объединения реализуется с помощью следующих конструкций: конструктор, распознаватель и поле объединения. Все они являются ПЕРВИЧНЫМИ-ВЫРАЖЕНИЯМИ (см. разд. 6).

КОНСТРУКТОР ::= ИМЯ-КОНСТРУКТОРА [ ( СПИСОК-ВЫРАЖЕНИЙ ) ]

Выражения, подставляемые в качестве аргументов КОНСТРУКТОРА, по их числу и типам должны соответствовать ОПИСАНИЯМ-ПОЛЕЙ в соответствующем ОПИСАНИИ-КОНСТРУКТОРА. Правила соответствия такие же, как для вызова предиката. Значением конструкции КОНСТРУКТОР является имя конструктора вместе со значениями его аргументов, если аргументы присутствуют.

РАСПОЗНАВАТЕЛЬ ::= ИМЯ-РАСПОЗНАВАТЕЛЯ ( ВЫРАЖЕНИЕ )

ИМЯ-РАСПОЗНАВАТЕЛЯ ::= ИМЯ-КОНСТРУКТОРА ?

Символ “?” является частью имени распознавателя и пишется слитно с именем конструктора, без пробела. ВЫРАЖЕНИЕ имеет тип объединения. Распознаватель является функцией типа bool. Значение РАСПОЗНАВАТЕЛЯ истинно, если значением ВЫРАЖЕНИЯ является конструктор с именем ИМЯ-КОНСТРУКТОРА.

ПОЛЕ-ОБЪЕДИНЕНИЯ ::= ПЕРВИЧНОЕ-ВЫРАЖЕНИЕ . ИМЯ-ПОЛЯ

Значением ПЕРВИЧНОГО-ВЫРАЖЕНИЯ является переменная типа “объединение”. Значением ПОЛЯ-ОБЪЕДИНЕНИЯ является значение поля с именем ИМЯ-ПОЛЯ в структуре переменной типа “объединение”. Использование конструкции ПОЛЕ-ОБЪЕДИНЕНИЯ корректно лишь в случае истинности распознавателя ИМЯ-КОНСТРУКТОРА?( ПЕРВИЧНОЕ-ВЫРАЖЕНИЕ ) для конструктора, к которому относится поле с именем ИМЯ-ПОЛЯ, в противном случае исполнение конструкции не определено.

Имена конструкторов и распознавателей являются глобально определенными в теле модуля, содержащего ИЗОБРАЖЕНИЕ-ТИПА-ОБЪЕДИНЕНИЯ.

ОПРЕДЕЛЕНИЕ-КОНСТРУКТОРА ::=

ИМЯ-КОНСТРУКТОРА [ ( ОПРЕДЕЛЕНИЕ-ПОЛЕЙ-КОНСТРУКТОРА ) ]  
 ОПРЕДЕЛЕНИЕ-ПОЛЕЙ-КОНСТРУКТОРА ::= ОПРЕДЕЛЕНИЕ-ПОЛЯ-КОНСТРУКТОРА  
 [ , ОПРЕДЕЛЕНИЕ-ПОЛЕЙ-КОНСТРУКТОРА ]

ОПРЕДЕЛЕНИЕ-ПОЛЯ-КОНСТРУКТОРА ::=

[ИЗОБРАЖЕНИЕ-ТИПА-ПЕРЕМЕННОЙ] ИДЕНТИФИКАТОР

Конструкция ОПРЕДЕЛЕНИЕ-КОНСТРУКТОРА используется в качестве АЛЬТЕРНАТИВЫ после case в операторе выбора (см. разд. 5) для выражения типа объединения. Значением АЛЬТЕРНАТИВЫ является конструктор с именем ИМЯ-КОНСТРУКТОРА и набором переменных, обозначающих поля данного конструктора. Эти переменные являются локальными в ОПЕРАТОРЕ-АЛЬТЕРНАТИВЫ, исполняемом для данной АЛЬТЕРНАТИВЫ. Типы переменных могут не указываться, поскольку они определяются из ИЗОБРАЖЕНИЯ-ТИПА-ОБЪЕДИНЕНИЯ.

```

1 type int_tree = union (
2   leaf (int val),
3   node (int_tree lhs, int v, int_tree rhs)
4 );
5
6 int_tree foo;
7 // ...
8 switch (foo) {
9   case leaf(v0): print("leaf", v0)
10  case node(l, v1, r): print("node with ", v1)
11 };
12
13 int_tree one_leaf = leaf(42);
14 int_tree small_tree = node(leaf(1), 2,
15                             node(leaf(3), 4, leaf(5)));

```

```

16 // small_tree задаёт следующее дерево:
17 //      2
18 //     / \
19 //    1   4
20 //     / \
21 //    3   5

```

Пример 10. Использование объединений

ИЗОБРАЖЕНИЕ-ПЕРЕЧИСЛЕНИЯ ::= **enum** ( ИДЕНТИФИКАТОР ( , ИДЕНТИФИКАТОР)\* )

Тип перечисления является частным случаем типа объединения с конструкторами без аргументов. Описатель **enum** трактуется как эквивалент **union**.

```
type fruit = enum (apple , orange , banana);
```

Пример 11. Описание перечисления

Структура вида “список” представляется как упорядоченная совокупность однородных элементов. Тип списка из элементов типа **T** определяется следующим описанием:

```

type list (type T) = union (
    nil ,
    cons (T car , list (T) cdr)
);

```

Тип **list** считается определенным в языке **P** и не требует описания в программе. Имена **list**, **nil** и **cons** глобально определены в предикатной программе.

Конструктор **nil** определяет пустой список, не содержащий элементов. Остальные значения типа **list** представляются конструктором **cons(x, s)**, где элемент **x** есть первый элемент списка – голова списка, а **s** есть оставшаяся часть списка – хвост списка. Функция **len(s)** определяет число элементов списка **s**. Операция конкатенации **s1 + s2** определяет список, состоящий из элементов списка **s1**, за которыми следуют элементы списка **s2**. Аргументами конкатенации могут быть также элементы списка. Функция **last(s)** определяет последний элемент непустого списка **s**; функция **prec(s)** – оставшуюся часть списка без последнего элемента.

ЭЛЕМЕНТ-СПИСКА ::= ПЕРВИЧНОЕ-ВЫРАЖЕНИЕ [ ИНДЕКС-ЭЛЕМЕНТА ]

Данная конструкция определяет значение элемента с указанным индексом в списке. Значением ПЕРВИЧНОГО-ВЫРАЖЕНИЯ является список.

**ИНДЕКС-ЭЛЕМЕНТА ::= ВЫРАЖЕНИЕ**

Для списка **s** конструкция **s[i]** определяет **i**-й элемент. Допустимыми являются значения индекса **i** от 0 до **len(s)-1**.

```
1 type int_list = list (int);
2 int_list s    = [[ 1, 1, 2, 3, 5, 8 ]];
3 int s_car     = s.car;           // s_car = 1
4 int_list s_cdr = s.cdr;
5              // s_cdr = [[ 1, 2, 3, 5, 8 ]]
6 int_list ss   = s + [[ 13, 21 ]];
7              // ss = [[ 1, 1, 2, 3, 5, 8, 13, 21 ]]
8 int count     = len (s);         // count = 6
9 int e         = s[3];           // e = 3
```

Пример 12. Операции со списком

**СТРОКОВЫЙ-ТИП ::= string**

Строковый тип **string** является обозначением типа списка **list(char)**. В дополнение к определенным выше операциям со списками имеется возможность задания конкретного значения строкового типа в виде строковой константы (см. разд. 2).

**ИЗОБРАЖЕНИЕ-ТИПА-МНОЖЕСТВА ::= set ( ИЗОБРАЖЕНИЕ-БАЗОВОГО-ТИПА )**

**ИЗОБРАЖЕНИЕ-БАЗОВОГО-ТИПА ::= ИЗОБРАЖЕНИЕ-ТИПА**

Тип множества есть множество всех подмножеств некоторого конечного базового типа. Имеется унарная операция **~** дополнения множества. Определены: операция **“+”** объединения множеств, операция **“-”** разности множеств, а также вычитания элемента из множества, операция **&** пересечения множеств, операция **or** объединения множеств, операция **in** принадлежности элемента множеству (см. разд. 6). Число элементов в множестве реализуется функцией **len**.

Операция **mask** преобразует множество в целое значение. Операция **bits** реализует обратную операцию преобразования целого во множество.

```
1 type int_set_t = set (1..1000);
2 int_set_t pow_2 = { 0, 2, 4, 8, 16, 32 };
3 bool is_pow_2   = 4 in pow_2; // is_pow_2 = true
4 int_set_t more_pow_2 = pow_2 + { 64, 128 };
```

Пример 13. Использование множеств

## 8. МАССИВЫ

### 8.1. Описание типа массива

```
ИЗОБРАЖЕНИЕ-ТИПА-МАССИВА ::=
    array ( ИЗОБРАЖЕНИЕ-ТИПА-ЭЛЕМЕНТА , ИЗМЕРЕНИЯ-МАССИВА )
ИЗОБРАЖЕНИЕ-ТИПА-ЭЛЕМЕНТА ::= ИЗОБРАЖЕНИЕ-ТИПА
ИЗМЕРЕНИЯ-МАССИВА ::=
    ИЗОБРАЖЕНИЕ-ТИПА-ИНДЕКСА [ , ИЗМЕРЕНИЯ-МАССИВА ]
ИЗОБРАЖЕНИЕ-ТИПА-ИНДЕКСА ::= ИЗОБРАЖЕНИЕ-ТИПА
```

Значение массива состоит из совокупности элементов. Каждый элемент доступен по набору индексов в массиве: для массива  $A$  элемент с набором индексов  $m$  есть  $A[m]$ . Число индексов совпадает с числом измерений массива. Тип каждого индекса определяет совокупность допустимых значений индекса и должен быть конечным.

Имеется операция “+” для объединения двух массивов-операндов в один массив при условии, что типы элементов совпадают, а множества наборов индексов операндов совместимы по типам и не пересекаются.

```
1 type int_vec = array (int , 1..5);
2 type int_mtx = array (int , 1..2 , 1..3);
3
4 int_vec v = [ 3, 1, 4, 1, 5 ];
5 int_vec w = for (x) x*x;
6 int v_2 = v [2]; // v_2 = 1
7
8 int_mtx m = [ [ 1, 2, 3 ], [ 104, 5, 6 ] ];
9 int e = m [2, 3]; // e = 6
```

Пример 14. Описание массивов и их инициализация

### 8.2. Вырезка массива

```
ВЫРЕЗКА-МАССИВА ::= ВЫРАЖЕНИЕ-МАССИВ
    [ СУЖЕННЫЙ-НАБОР-ТИПОВ-ИНДЕКСОВ ]
ВЫРАЖЕНИЕ-МАССИВ ::= ВЫРАЖЕНИЕ
СУЖЕННЫЙ-НАБОР-ТИПОВ-ИНДЕКСОВ ::= НАБОР-ИНДЕКСОВ
НАБОР-ИНДЕКСОВ ::= ЭЛЕМЕНТ-НАБОРА-ИНДЕКСОВ [ , НАБОР-ИНДЕКСОВ ]
ЭЛЕМЕНТ-НАБОРА-ИНДЕКСОВ ::= ЗНАЧЕНИЕ-ИНДЕКСА | ИЗОБРАЖЕНИЕ-ДИАПАЗОНА
ЗНАЧЕНИЕ-ИНДЕКСА ::= ВЫРАЖЕНИЕ
```

Вырезка массива определяет массив как часть некоторого массива заданием подмножества наборов индексов. Достаточно задать сужение

хотя бы по одному измерению массива. В частности допустимо сужение к единственному значению индекса. Однако вырезка массива не может превратиться в единственный элемент массива. Результатом исполнения вырезки массива является переменная типа “массив”. Нового массива при этом не создается.

### 8.3. Определение массива

```
ОПРЕДЕЛЕНИЕ-МАССИВА ::=
    ОПРЕДЕЛЕНИЕ-ИНДЕКСОВ ОПРЕДЕЛЕНИЕ-ЭЛЕМЕНТА-МАССИВА |
    АГРЕГАТ-МАССИВ |
    ОПРЕДЕЛЕНИЕ-МАССИВА-ПО-ЧАСТЯМ
ОПРЕДЕЛЕНИЕ-МАССИВА-ПО-ЧАСТЯМ ::=
    ОПРЕДЕЛЕНИЕ-ИНДЕКСОВ ОПРЕДЕЛЕНИЕ-ЧАСТЕЙ-МАССИВА
```

Результатом вычисления ОПРЕДЕЛЕНИЯ-МАССИВА является значение массива. Вычисление реализуется итерацией по всевозможным значениям набора индексов массива. Для каждого набора индексов вычисляется соответствующий элемент массива. Вычисление значений разных элементов может проводиться параллельно.

```
ОПРЕДЕЛЕНИЕ-ЭЛЕМЕНТА-МАССИВА ::= ВЫРАЖЕНИЕ
```

ВЫРАЖЕНИЕ для элемента массива зависит от индексов, указанных в ОПРЕДЕЛЕНИИ-ИНДЕКСОВ.

```
ОПРЕДЕЛЕНИЕ-ИНДЕКСОВ ::= for ( ЗАДАНИЕ-ИНДЕКСОВ )
ЗАДАНИЕ-ИНДЕКСОВ ::= ОПРЕДЕЛЕНИЕ-ИНДЕКСА [ , ЗАДАНИЕ-ИНДЕКСОВ ]
ОПРЕДЕЛЕНИЕ-ИНДЕКСА ::= [ИЗОБРАЖЕНИЕ-ТИПА-ПЕРЕМЕННОЙ] ИНДЕКС
ИНДЕКС ::= ИДЕНТИФИКАТОР
```

Переменные, обозначающие индексы, локальны в ОПРЕДЕЛЕНИИ-МАССИВА. При определении индекса обычно используется описатель `var` (см. разд. 4). Указание типа индекса требуется в редких случаях, когда тип массива (в том числе и типы индексов массива) трудно определить из позиции, в которой находится ОПРЕДЕЛЕНИЕ-МАССИВА. Отметим, что в ОПРЕДЕЛЕНИИ-ИНДЕКСА тип индекса, если он задан явно, должен точно покрывать множество значений индекса; обычно это диапазон типа `int`.

```
1 type ar1_5 = array (int , 1..5);
2 ar1_5 squ;
3 squ = for (var i) i*i;
4 ar1_5 r = for (var i) 100 - i;
```

Пример 15. Определения массивов

```

ОПРЕДЕЛЕНИЕ-ЧАСТЕЙ-МАССИВА ::=
  { (ОПРЕДЕЛЕНИЕ-ЧАСТИ-МАССИВА)+
    [default : ОПРЕДЕЛЕНИЕ-ЭЛЕМЕНТА-МАССИВА ] }
ОПРЕДЕЛЕНИЕ-ЧАСТИ-МАССИВА ::=
  case ИНДЕКСЫ-ЧАСТИ : ОПРЕДЕЛЕНИЕ-ЭЛЕМЕНТА-МАССИВА
ИНДЕКСЫ-ЧАСТИ ::= ЭЛЕМЕНТ-НАБОРА-ИНДЕКСОВ | ( НАБОР-ИНДЕКСОВ ) |
  ( ЭЛЕМЕНТ-НАБОРА-ИНДЕКСОВ [, ИНДЕКСЫ-ЧАСТИ] ) |
  ( ( НАБОР-ИНДЕКСОВ ) [, ИНДЕКСЫ-ЧАСТИ] )

```

ИНДЕКСЫ-ЧАСТИ определяют некоторое подмножество на произведении типов индексов. Эти подмножества не должны пересекаться для разных ОПРЕДЕЛЕНИЙ-ЧАСТИ-МАССИВА. Определение элементов массива для наборов индексов, не принадлежащих ни одной из указанных частей массива, реализуется частью `default`. При отсутствии части `default` объединение подмножеств наборов индексов для разных частей должно совпадать с полным множеством наборов индексов. Вычисление элементов массива по каждой из частей массива проводится независимо, возможно параллельно.

Если ОПРЕДЕЛЕНИЕ-ЧАСТЕЙ-МАССИВА используется в операции модификации массива (см. разд. 6), то часть `default` отсутствует.

```

1 type Ar(nat k) = array (real, 1..k);
2 F (nat n, Ar(n) x: Ar(n+1) x')
3 { x' = for (var j) { case 1..n : x[j] + 1 case n + 1 : 0 } }

```

Пример 16. Определение массива по частям

```

1 type MATR(nat k) = array(real, 1..k, 1..k);
2 perm_lines(nat n, MATR(n) a, nat k,m : MATR(n) a')
3 pre 1 <= k < m <= n
4 {
5     a' = a with for (var i, j) {
6         case (k, 1..n): a[m, j]
7         case (m, 1..n): a[k, j]
8     }
9 }

```

Пример 17. Перестановка двух строк матрицы

В матрице **a** переставляются местами строки с номерами **k** и **m**. Перестановка реализуется операцией модификации (см. разд. 6) двух строк в массиве **a**; остальные строки остаются неизменными. Задание нового значения двух строк матрицы реализуется конструкцией ОПРЕДЕЛЕНИЕ-ЧАСТЕЙ-МАССИВА.

## 8.4. Объединение массивов

Операндами операции “+” объединения массивов

**ВЫРАЖЕНИЕ-МАССИВ + ВЫРАЖЕНИЕ-МАССИВ**

являются выражения, значениями которых являются массивы. Объединяемые массивы должны иметь совпадающие типы элементов и непересекающиеся типы индексов, причем типы индексов должны принадлежать некоторому общему типу. Результатом операции является массив. Тип индексов результирующего массива есть объединение типов индексов операндов, а тип элементов тот же, что и у операндов. Произвольный элемент результирующего массива есть либо элемент первого массива-операнда, либо элемент второго.

## 9. ФОРМУЛЫ

Формулы, используемые в качестве предусловий и постусловий предикатов, есть формулы типизированного исчисления предикатов высших порядков. Подформула, входящая в предусловие или постусловие, может быть определена отдельно в виде описания формулы.

**ОПИСАНИЕ-ФОРМУЛЫ ::=**

**formula** ИМЯ-ФОРМУЛЫ ( ОПИСАНИЯ-АРГУМЕНТОВ [ : ТИП-РЕЗУЛЬТАТА ] )  
= ФОРМУЛА

ИМЯ-ФОРМУЛЫ ::= ИДЕНТИФИКАТОР

ТИП-РЕЗУЛЬТАТА ::= ИЗОБРАЖЕНИЕ-ТИПА

**ОПИСАНИЕ-ФОРМУЛЫ** вводит имя формулы для обозначения выражения, представленного **ФОРМУЛОЙ**. Переменные, входящие в **ФОРМУЛУ**, должны быть указаны в **ОПИСАНИЯХ-АРГУМЕНТОВ** (см. разд. 3.1). Описание формулы помещается среди описаний модуля программы (см. разд. 4). **ФОРМУЛА** является выражением типа **ТИП-РЕЗУЛЬТАТА**. Если **ТИП-РЕЗУЛЬТАТА** не указан, то **ФОРМУЛА** имеет тип **bool**.

Использование формулы, определенной **ОПИСАНИЕМ-ФОРМУЛЫ**, в других формулах реализуется через вызов формулы.

**ВЫЗОВ-ФОРМУЛЫ ::= ИМЯ-ФОРМУЛЫ ( СПИСОК-ВЫРАЖЕНИЙ )**

Описание формулы может быть рекурсивным: **ФОРМУЛА** в правой части описания может содержать рекурсивный вызов этой формулы. Не допускается взаимной рекурсии в описаниях двух разных формул. Другое требование: рекурсивный вызов должен находиться в позитивной позиции **ФОРМУЛЫ**. Понятия позитивной и негативной позиции определены ниже.

ФОРМУЛА ::= ВЫРАЖЕНИЕ | ( ФОРМУЛА ) | ВЫЗОВ-ФОРМУЛЫ |  
           ! ФОРМУЛА | ФОРМУЛА & ФОРМУЛА | ФОРМУЛА or ФОРМУЛА |  
           ФОРМУЛА => ФОРМУЛА | ФОРМУЛА <=> ФОРМУЛА |  
           КВАНТОРНАЯ-ЧАСТЬ ФОРМУЛА

Все альтернативы приведенного правила, кроме первых трех, определяют формулы типа **bool**, т.е. предикаты. Операции “!”, “&” и “or” имеют тот же смысл, что и для логических выражений. Операция “=>” обозначает импликацию, “<=>” – логическое тождество.

КВАНТОРНАЯ-ЧАСТЬ ::= =

          КВАНТОР СПИСОК-ПОДКВАНТОРНЫХ-ПЕРЕМЕННЫХ . [КВАНТОРНАЯ-ЧАСТЬ]

КВАНТОР ::= КВАНТОР-ВСЕОБЩНОСТИ | КВАНТОР-СУЩЕСТВОВАНИЯ

КВАНТОР-ВСЕОБЩНОСТИ ::= ! | **forall**

КВАНТОР-СУЩЕСТВОВАНИЯ ::= ? | **exists**

Использование **forall** вместо “!” предпочтительно при вхождении квантора внутри формулы, поскольку “!” ассоциируется также с отрицанием.

СПИСОК-ПОДКВАНТОРНЫХ-ПЕРЕМЕННЫХ ::=

          ИЗОБРАЖЕНИЕ-ТИПА ПОДКВАНТОРНАЯ-ПЕРЕМЕННАЯ

          ( , ПОДКВАНТОРНАЯ-ПЕРЕМЕННАЯ ) \*

          [ , СПИСОК-ПОДКВАНТОРНЫХ-ПЕРЕМЕННЫХ ]

ПОДКВАНТОРНАЯ-ПЕРЕМЕННАЯ            ::= ИДЕНТИФИКАТОР

Приоритеты логических связок в порядке убывания следующие: !, &, =>, <=>, кванторы !, **forall**, ? и **exists**.

Определим понятие позиции, позитивной или негативной, для произвольного вхождения подформулы. Позиция ФОРМУЛЫ в качестве правой части ОПИСАНИЯ-ФОРМУЛЫ является позитивной. Допустим, формула X есть одна из следующих формул: A & B, A or B, !C, C => A, !y.A, ?y.A. Для перечисленных случаев подформулы A и B имеют ту же позицию, что формула X, а подформула C меняет позицию на противоположную. Все остальные позиции подформул считаются неизвестными.

## 10. ПРОЦЕССЫ

Для большинства программ можно построить спецификацию в виде предусловия и постусловия. Однако не всякую программу можно специфицировать таким образом. Имеется класс программ реального времени, в частности, реактивных систем, где программа представлена в виде набора параллельных взаимодействующих процессов. Примерами таких программ являются протоколы и телекоммуникационные си-

стемы. В целях спецификации и реализации данного класса программ язык P расширен средствами для описания процессов, передачи сообщений и порождения процессов, работающих параллельно с процессом-родителем.

**ОПРЕДЕЛЕНИЕ-ПРОЦЕССА ::=**

**process** ИМЯ-ПРОЦЕССА [ ( [ОПИСАНИЯ-АРГУМЕНТОВ] :  
 ОПИСАНИЯ-РЕЗУЛЬТАТОВ-ГИПЕРФУНКЦИИ ) ]  
 { ТЕЛО-ПРОЦЕССА [МЕТКА :] }  
 [spec ВНЕШНЯЯ-СПЕЦИФИКАЦИЯ-ПРОЦЕССА]

ИМЯ-ПРОЦЕССА ::= ИДЕНТИФИКАТОР

Процесс может быть бесконечным, и в этом случае он не имеет результатов. Результаты конечного завершающегося процесса определяются так же, как у гиперфункции. Конструкция вида **имя:** в теле процесса является описанием локальной метки, на которую возможен переход из операторов процесса. Определяемый процесс имеет структуру машины конечных состояний в виде гиперграфа, в котором гипердугами являются подпроцессы или гиперфункции. Использование аппарата гиперфункций обеспечивает большую гибкость в декомпозиции процессов.

**ВНЕШНЯЯ-СПЕЦИФИКАЦИЯ-ПРОЦЕССА** описывает эффект функционирования процесса на внешнее окружение, в рамках которого существует процесс. В общем случае этот эффект невозможно выразить с помощью предусловия и постусловия. В качестве языков внешней спецификации процессов ранее использовались язык алгебры процессов CCS [8] Р. Милнера и язык спецификации TLA+ [9] Л. Лампорта. Язык внешней спецификации в данном документе не фиксируется. Наиболее адекватным и проработанным представляется язык SAL [10].

**ТЕЛО-ПРОЦЕССА ::= ОПИСАНИЕ-ПЕРЕМЕННЫХ [ ; ТЕЛО-ПРОЦЕССА ] |**  
**ВОЗМОЖНО-ПОМЕЧЕННЫЙ-ОПЕРАТОР [ ; ТЕЛО-ПРОЦЕССА ]**  
**ВОЗМОЖНО-ПОМЕЧЕННЫЙ-ОПЕРАТОР ::= [МЕТКА :] ОПЕРАТОР**

Переменные и метки, описанные в теле процесса, считаются локальными в данном процессе. Описания и операторы блока исполняются в порядке их следования. Оператор перехода изменяет последовательность выполнения. Допускаются циклы в теле процесса.

Языковые конструкции **ОПЕРАТОР**, **ПЕРВИЧНОЕ-ВЫРАЖЕНИЕ** и **ИДЕНТИФИКАЦИЯ-ПРЕДИКАТА**, определенные выше, имеют доопределение для процессов.

**ИДЕНТИФИКАЦИЯ-ПРЕДИКАТА ::= ... | [ОБЪЕКТ-КЛАССА .] ИМЯ-ПРЕДИКАТА**  
**ОБЪЕКТ-КЛАССА ::= ПЕРВИЧНОЕ-ВЫРАЖЕНИЕ**

Конструкция ИДЕНТИФИКАЦИЯ-ПРЕДИКАТА используется в вызове предиката для указания вызываемого предиката.

В теле процесса наряду с операторами, разрешенными в определении предиката, используются дополнительные операторы.

```
ОПЕРАТОР-РАБОТЫ-С-ПРОЦЕССАМИ ::=
    ВЫЗОВ-ПРОЦЕССА | ЗАПУСК-НЕЗАВИСИМОГО-ПРОЦЕССА |
    ПАРАЛЛЕЛЬНАЯ-КОМПОЗИЦИЯ-ПРОЦЕССОВ | ПРИЕМ-СООБЩЕНИЯ |
    ПОСЫЛКА-СООБЩЕНИЯ | ЗАЩИЩЕННЫЙ-ОПЕРАТОР | ОПЕРАТОР-ПЕРЕХОДА
ВЫЗОВ-ПРОЦЕССА ::= [ ОБЪЕКТ-КЛАССА . ] ИМЯ-ПРОЦЕССА
    ( [ АРГУМЕНТЫ ] ( : РЕЗУЛЬТАТЫ-ВЕТВИ)* ) )
```

Процесс, запускаемый на исполнение ВЫЗОВОМ-ПРОЦЕССА, должен иметь соответствующее определение процесса. Если определение процесса принадлежит некоторому классу, то вызов процесса реализуется через ОБЪЕКТ-КЛАССА.

```
ПАРАЛЛЕЛЬНАЯ-КОМПОЗИЦИЯ-ПРОЦЕССОВ ::= ПАРАЛЛЕЛЬНЫЙ-ОПЕРАТОР
```

По синтаксису параллельная композиция процессов аналогична параллельному оператору. Параллельная композиция задает параллельное исполнение процессов, вызовы которых указаны в композиции. Взаимодействие между процессами реализуется посредством приема и передачи сообщений.

```
ОПИСАНИЕ-СООБЩЕНИЯ ::= message ИМЯ-СООБЩЕНИЯ [( СПИСОК-ТИПОВ ) ] |
    queue ИМЯ-СООБЩЕНИЯ [( СПИСОК-ТИПОВ ) ]
ИМЯ-СООБЩЕНИЯ ::= ИДЕНТИФИКАТОР
СПИСОК-ТИПОВ ::= ИЗОБРАЖЕНИЕ-ТИПА [ , СПИСОК-ТИПОВ ]
```

Взаимодействие процесса с окружением реализуется посредством сообщений. Сообщение содержит (возможно, пустой) набор значений, типы которых указываются в параметрах описания сообщения.

```
ПОСЫЛКА-СООБЩЕНИЯ ::=
    ИМЯ-ПРОЦЕССА ! ИМЯ-СООБЩЕНИЯ [( СПИСОК-ВЫРАЖЕНИЙ ) ] |
    ИМЯ-СООБЩЕНИЯ ! [( СПИСОК-ВЫРАЖЕНИЙ ) ]
```

Оператор посылает сообщение с набором значений, вычисленных СПИСОКОМ-ВЫРАЖЕНИЙ. Если сообщение определено описанием **message**, то выполнение оператора не будет завершено до тех пор, пока это сообщение не будет принято получателем. Для сообщения, определенного описанием **queue**, оператор вставляет сообщение в конец очереди посланных сообщений и завершает свою работу.

```
ПРИЕМ-СООБЩЕНИЯ ::= receive СООБЩЕНИЕ | ПРИЕМ-СООБЩЕНИЯ-ОБЩЕГО-ВИДА
СООБЩЕНИЕ ::= ИМЯ-СООБЩЕНИЯ [( СПИСОК-ПЕРЕМЕННЫХ ) ]
```

Допустим, что процесс, исполняющий оператор `receive` СООБЩЕНИЕ, получил сообщение с именем, указанным в операторе. Исполнение оператора заключается в присваивании перечисленным переменным соответствующих значений, поставляемых сообщением. Если исполняемый процесс не получил сообщения с указанным именем, то исполнение оператора приостанавливается (блокируется) до поступления требуемого сообщения.

```
ПРИЕМ-СООБЩЕНИЯ-ОБЩЕГО-ВИДА ::= receive { СООБЩЕНИЕ : ОПЕРАТОР
    ( case СООБЩЕНИЕ : ОПЕРАТОР ) *
    [ after ВРЕМЯ-ОЖИДАНИЯ : ОПЕРАТОР ]
  }
```

```
ВРЕМЯ-ОЖИДАНИЯ ::= ВЫРАЖЕНИЕ
```

В результате выполнения оператора общего вида реализуется прием одного из указанных сообщений: того, которое приходит первым. При этом соответствующим переменным присваиваются значения параметров сообщения. Далее исполняется ОПЕРАТОР, следующий за ":" после пришедшего сообщения. Оператор общего вида также является блокирующим: исполнение текущего процесса приостанавливается до поступления любого из указанных сообщений. Если в операторе присутствует подконструкция `after`, то ожидание поступления одного из перечисленных сообщений ограничено по времени. Время ожидания имеет тип `nat` и определяет значение времени в миллисекундах, по истечении которого срабатывает ОПЕРАТОР, указанный завершителем `after`.

```
ПЕРВИЧНОЕ-ВЫРАЖЕНИЕ ::= ... | СООБЩЕНИЕ
```

Конструкция СООБЩЕНИЕ в любой позиции логического выражения трактуется как неблокирующий прием сообщений. Исполнение конструкции осуществляется следующим образом. В очереди пришедших сообщений исполняемого процесса ищется сообщение с именем, указанным в конструкции СООБЩЕНИЕ. Если сообщение обнаружено, оно вычеркивается из очереди. При этом значения параметров присваиваются соответствующим переменным, перечисленным в круглых скобках. Результатом исполнения конструкции СООБЩЕНИЕ является значение `true`. Если требуемое сообщение не обнаружено в очереди, исполнение конструкции СООБЩЕНИЕ завершается значением `false`.

Динамическое создание нового процесса и его запуск реализуется через объект класса. Объект (или значение) типа "класс" является структурой, составленной из полей класса. Описание класса определяется следующим образом:

```

ОПРЕДЕЛЕНИЕ-КЛАССА ::=
    class ИМЯ-КЛАССА [extends ИМЯ-СУПЕРКЛАССА] { ТЕЛО-КЛАССА }
ИМЯ-КЛАССА ::= ИДЕНТИФИКАТОР
ТЕЛО-КЛАССА ::= ОПИСАНИЕ-КЛАССА [; ТЕЛО-КЛАССА]
ОПИСАНИЕ-КЛАССА ::=
    ОПИСАНИЕ-ТИПА | ОПИСАНИЕ-ПЕРЕМЕННЫХ | ОПИСАНИЕ-СООБЩЕНИЯ |
    ОПРЕДЕЛЕНИЕ-ПРОЦЕССА | ОПРЕДЕЛЕНИЕ-ПРЕДИКАТА |
    ОПИСАНИЕ-КОНСТРУКТОРА-КЛАССА
ОПИСАНИЕ-КОНСТРУКТОРА-КЛАССА ::=
    ИМЯ-КЛАССА ([ОПИСАНИЯ-АРГУМЕНТОВ]) { ТЕЛО-КОНСТРУКТОРА }
ТЕЛО-КОНСТРУКТОРА ::= [ ТЕЛО-БЛОКА ]

```

Доступ к элементу класса вне описания класса реализуется через объект класса, т.е. конструкцией `объект.имя`, где `имя` – имя элемента класса. Доступ к элементу класса в теле класса осуществляется непосредственно по имени элемента. Для обозначения объекта, определяемого данным классом, используется имя `this`.

Отметим, что классы не интегрированы с системой типов языка Р.

```

ПЕРВИЧНОЕ-ВЫРАЖЕНИЕ ::= ... | СОЗДАНИЕ-ОБЪЕКТА
СОЗДАНИЕ-ОБЪЕКТА ::= new ИМЯ-КЛАССА ( [АРГУМЕНТЫ] )

```

Операция СОЗДАНИЯ-ОБЪЕКТА реализует вызов конструктора, параметры которого соответствуют списку аргументов. Результатом операции является новый объект (значение) типа ИМЯ-КЛАССА. При исполнении конструктора создается объект – структура, полями которой являются переменные, описанные в теле класса. Для созданного объекта исполняются ОПИСАНИЯ-ПЕРЕМЕННЫХ, находящиеся в теле класса и содержащие инициализацию переменных. Далее исполняется тело конструктора.

Если для операции `new ИМЯ-КЛАССА( )` нет соответствующего конструктора с пустым списком аргументов, то такой конструктор считается определенным, и его тело пусто.

```

ИМЯ-СУПЕРКЛАССА ::= ИМЯ-КЛАССА

```

При наличии подконструкции `extends ИМЯ-СУПЕРКЛАССА` определяемый класс является расширением суперкласса, представленного описателем `extends`. Структура, создаваемая для определяемого класса, содержит поля суперкласса, за которыми следуют поля определяемого класса. Конструктор класса содержит явный или неявный вызов соответствующего конструктора для суперкласса.

```

ПЕРВИЧНОЕ-ВЫРАЖЕНИЕ ::= ... | ДИНАМИЧЕСКОЕ-СОЗДАНИЕ-ПРОЦЕССА
ДИНАМИЧЕСКОЕ-СОЗДАНИЕ-ПРОЦЕССА ::= new ИМЯ-ПРОЦЕССА ( [<АРГУМЕНТЫ>] )

```

Данная операция реализует создание нового независимого процесса и его запуск на исполнение. Процесс с указанным именем должен быть элементом некоторого класса. В начале тела процесса находится явный или неявный вызов конструктора класса. При исполнении операции **new** срабатывает вызов конструктора в начале тела процесса. Результатом вызова конструктора является новый объект класса. Далее создается новый процесс и запускается на исполнение параллельно с текущим исполняемым процессом. Программой созданного процесса является тело процесса, а глобальными данными – поля созданного объекта. На этом исполнение операции **new** завершается. Ее результатом является созданный объект. Текущий процесс продолжает исполняться параллельно с созданным процессом. Объект, являющийся результатом операции **new**, используется в качестве указателя созданного процесса в операторах обмена сообщениями.

**ЗАЩИЩЕННЫЙ-ОПЕРАТОР ::=**  
**with ( СПИСОК-РАЗДЕЛЯЕМЫХ-ПЕРЕМЕННЫХ ) ОПЕРАТОР**  
**СПИСОК-РАЗДЕЛЯЕМЫХ-ПЕРЕМЕННЫХ ::= СПИСОК-ПЕРЕМЕННЫХ**

Взаимодействие между параллельными процессами возможно через разделяемые переменные, доступные для нескольких процессов и являющиеся глобальными по отношению к ним. Доступ к разделяемым переменным разрешен только внутри защищенного оператора. При исполнении ОПЕРАТОРА доступ к перечисленным разделяемым переменным в других процессах блокируется. Блокировка снимается при завершении исполнения ОПЕРАТОРА. Если при попытке исполнить ЗАЩИЩЕННЫЙ-ОПЕРАТОР одна из разделяемых переменных оказалась заблокирована другим процессом, исполнение текущего процесса, исполняющего ЗАЩИЩЕННЫЙ-ОПЕРАТОР, приостанавливается до момента снятия блокировки с разделяемой переменной.

## 11. ИМПЕРАТИВНОЕ РАСШИРЕНИЕ

Императивное расширение языка Р определяет дополнительные языковые конструкции, возникающие в программе в результате проведения трансформаций предикатной программы; см. введение. Использование этих конструкций в исходной программе недопустимо.

**ЗАГОЛОВОК-ПРЕДИКАТА ::=**  
**( [ОПИСАНИЯ-АРГУМЕНТОВ] [: ОПИСАНИЯ-РЕЗУЛЬТАТОВ] )**  
**ВЫЗОВ-ПРЕДИКАТА-ФУНКЦИИ ::=**  
**ИДЕНТИФИКАЦИЯ-ПРЕДИКАТА ( [АРГУМЕНТЫ] [: РЕЗУЛЬТАТЫ] )**

В императивном расширении для предиката-функции допускается отсутствие результатов. Это отражено в приведенных определениях, расширяющие определения, данные в разд. 3.1. Отсутствие результатов возможно, например, после склеивания результатов с глобальными переменными.

Определим дополнительные операторы императивного расширения.

ОПЕРАТОР-ИМПЕРАТИВНОГО-РАСШИРЕНИЯ ::=

**break** | ОПЕРАТОР-FOR | ОПЕРАТОР-ЕСЛИ

В императивной программе, полученной в результате трансформации, в позиции оператора исходной предикатной программы может также находиться оператор императивного расширения. Отметим, что групповой оператор присваивания, возникающий при подстановке тела предиката на место вызова предиката, является частным случаем оператора присваивания, когда в левой части – мультипеременная, а в правой – мультिवыражение.

```
ОПЕРАТОР-FOR      ::= for ( ЗАГОЛОВОК-ЦИКЛА ) { ОПЕРАТОР [; ОПЕРАТОР] }
ЗАГОЛОВОК-ЦИКЛА  ::= [[ИЗОБРАЖЕНИЕ-ТИПА] ПАРАМЕТР-ЦИКЛА = ВЫРАЖЕНИЕ] ;
                  [УСЛОВИЕ-ЗАВЕРШЕНИЯ] ;
                  [ПЕРЕСЧЕТ-ПАРАМЕТРА]
ПАРАМЕТР-ЦИКЛА   ::= ИДЕНТИФИКАТОР
УСЛОВИЕ-ЗАВЕРШЕНИЯ ::= ВЫРАЖЕНИЕ
ПЕРЕСЧЕТ-ПАРАМЕТРА ::= ОПЕРАТОР
```

В императивном расширении используются также операторы перехода и помеченные операторы. Синтаксис и семантика определены в разд. 10.

Семантика цикла **for** соответствует языку C++. Выход из цикла **for** может также быть реализован оператором **break** из тела цикла.

ЦИКЛ-WHILE ::= **while** ( ВЫРАЖЕНИЕ ) ОПЕРАТОР

Тело цикла **while** исполняется, пока логическое выражение в его заголовке истинно.

```

1 sum (list (int) a : int r) {
2   if (len (a) > 0) {
3     r = a [0];
4     int i = 1;
5     while (i < len (a)) {
6       r = r + a [i];
7       i = i + 1;
8     }
9   } else {
10    r = 0;
11  }
12 }

```

### Пример 18. Использование цикла `while`

Данная программа может получиться в результате применения первых трех трансформаций: склеивания переменных, замены хвостовой рекурсии циклом и подстановки определения предиката на место вызова.

Циклы `for` и `while` часто используются для представления циклов, реализуемых с помощью операторов перехода, в целях улучшения структуры программы.

## СПИСОК ЛИТЕРАТУРЫ

1. Шелехов В.И. Введение в предикатное программирование. – Новосибирск, 2002. – 82с. – (Препр. / ИСИ СО РАН; N 100).
2. Шелехов В.И. Язык предикатного программирования Р. – Новосибирск, 2002. – 40с. – (Препр. / ИСИ СО РАН; N 101).
3. Шелехов В.И. Предикатное программирование: основы, язык, технология. // Методы предикатного программирования / ИСИ СО РАН. – Новосибирск, 2003. – С.7-15.
4. Шелехов В.И. Разработка программы построения дерева суффиксов в технологии предикатного программирования. – Новосибирск, 2004. – 52с. – (Препр. / ИСИ СО РАН; N 115).
5. Шелехов В.И. Язык спецификации процессов // Методы предикатного программирования. Вып.2 / ИСИ СО РАН. – Новосибирск, 2006. – С. 17-34.
6. Шелехов В.И. Модель корректности программ на языке исчисления вычислимых предикатов. – Новосибирск, 2007. – 51с. – (Препр. / ИСИ СО РАН; N 145).
7. PVS Specification and Verification System. — <http://pvs.csl.sri.com/>
8. Milner R. A Calculus of Communicating Systems. // LNCS. Vol. 94. 1980.
9. Lamport L. Specifying concurrent systems with TLA+. — *Calculational System Design*, N. 173 in Series F: Computer and Systems Sciences, pages 183-247. — IOS Press, Amsterdam, 1999.
10. Moura L. , Owre S., and Shankar N. The SAL Language Manual. — SRI International. — <http://sal.csl.sri.com/doc/language-report.pdf>

**Н.С. Карнаухов, Д.Ю. Першин, В.И. Шелехов**

**ЯЗЫК ПРЕДИКАТНОГО ПРОГРАММИРОВАНИЯ Р**

**Препринт  
153**

Рукопись поступила в редакцию 11.02.2010

Рецензент А.Г. Марчук

Редактор Т. М. Бульонкова

---

Подписано в печать 13.05.2010

Формат бумаги 60×84 1/16

Объем 2,5 уч.-изд.л., 2,75 п.л.

Тираж 60 экз.

---

Центр оперативной печати “Оригинал 2”  
г.Бердск, ул. Островского, 55, оф. 02, тел. (383) 214 45 35